

4

מחשבת

גרפיקה עדינה



commodore64

משהו על חברת "מחשבת" – מערכות למידה

זו חברה, אשר אחד מתחומי התמחותה העיקריים הוא פיתוח מערכות לימוד עצמי.

מחברי יחידות לימוד אלה עוסקים במשך שנים בפיתוח מערכות ללימוד עצמי עבור המגזר האזרחי והצבאי.

חל איסור מוחלט להעתיק או לשכפל יחידת לימוד זו בשלמותה או בחלקה לכל מטרה שהיא או לעשות בה שימוש מסחרי כלשהו, ללא רשות בכתב מאת חברת מחשבת – מ.ל. בע"מ.

© 1986 – כל הזכויות שמורות למחשבת – מ.ל. בע"מ.

ת.ד. 48032 ת"א מיקוד 61480.

הוצאה לאור – מחשבת מ.ל. בע"מ.

דפוס: דפוס וכריכת מאירי

יהודה הלוי 4, תל־אביב, טלפון: 03-650455

Copyright © 1986 by MAKH-SHEVET M.L. Ltd.

P.O.Box 48032 Tel-Aviv 61480, Israel

Commodore 64



מחשבת

**צעדים מתקדמים במחשב
בלימוד עצמי**

יחידה 4
גרפיקה עדינה

(High-Resolution)

תוכן העניינים

עמוד

5	פתח דבר.....
6	פרק א - המסך הגרפי.....
11	פרק ב - קואורדינטות וצירים.....
18	פרק ג - כותבים במסך הגרפי.....
25	פרק ד - הצבעים נכנסים למסך הגרפי.....
33	פרק ה - עפרון וידאו עדין.....
39	פרק ו - מתיחת קוים על המסך.....
46	פרק ז - משפט פיתגורס.....
52	רשימת מושגים נלמדים וסיכום.....
53	נספח א - המספרים השליליים.....
58	נספח ב - סדר ביצוע פעולות חשבון.....
62	תשובות.....
74	רשת המסך הגרפי.....

פתח דבר

יחידת לימוד זאת הנקראית "גרפיקה עדינה" מוקדשת לפיתוח תוכניות בתחום הגרפיקה העדינה (HIGH RESOLUTION).

על מנת לעשות זאת עליך יהיה להשתמש בקסטה HIGH RESOLUTION, שפותחה במיוחד ע"י "מחשבת" עבור יחידת לימוד זו ועבור היחידות שיבואו אחריה. בעזרת הקסטה הרחבנו את ה-BASIC של הקומודור 64 גם לתחום חדש זה.

- עכשיו עליך להעלות את התוכנה אל תוך זכרוננו של המחשב (LOAD).
 - לאחר שהמסך עם הכתובת "מחשבת" מופיע - לחץ על מקש כלשהו - עכשיו המחשב מוכן לתחילת פעילות.
 - כדי לבדוק אם הקסטה הוטענה כשורה - פקוד על המחשב: `HELP (+) RETURN`.
- רשימה מלאה של הפקודות החדשות שנוספו למחשב מופיעה על המסך.

ביחידת לימוד זאת תלמד להשתמש בפקודות אלה על מנת לפתח תוכניות בגרפיקה העדינה. תוך כדי כך נעשיר גם את עולם המושגים המתמטי שלך על מנת לפתח תוכניות רב-גוניות.

טעות לעולם חוזרת

אנחנו מקווים כי לחוברת איך "בגים". אם בכל זאת תמצא כזה - נשמח אם תתקשה אלינו כך, שנוכל לתקן זאת במהדורה הבאה. בכל מקרה, אם תתקל בבעיה לא מובנת נשמח לעזור לך. צלצל ל-03-492723 ונען...

באיחולי המשך לימוד

מרתק ומהנה

דני קדם ויצחק קליסקי

מחשבת - מערכות למידה

פרק א

המסך הגרפי

GRAPHICS ו- PLOT, PIXEL

כתוב את השורה הבאה:

10 PLOT 150, 150

(את הפקודה PLOT תוכל לכתוב בקיצור P ו-L-SHIFT)

לأחר שהרצת את השורה (RUN) האם התקבל "משהו" על המסך? ?

לכאורה לא קרה כלום!

דע לך כי המחשב ביצע את הפקודה במסך אחר - "המסך הגרפי".
רוצה לראות?

לשם כך עליך לפקוד עליו לעבור למסך הגרפי:

GRAPHICS (גרפיקס) - "מסך - גרפי"

(גם את פקודת ה-GRAPHICS תוכל לכתוב בקיצור: G ו-R-SHIFT)

אם פעלת נכון אתה אמור לראות נקודה לבנה במסך הגרפי!

שים לב! כאשר אתה עובר למסך הגרפי, השוליים משנים את צבעם
לצבע חום-בהיר!

נסכם:

על-מנת לראות את המסך הגרפי יש תמיד לפקוד על המחשב לעבור

אליו בעזרת פקודת ה-GRAPHICS.

בעזרת פקודת ה-PLOT (פלוט) שפירושה: "צייר נקודה" יכול המחשב

לרשום בקודות קטנות במסך הגרפי.

נקודה כמו זו המופיעה עתה על המסך הגרפי היא הנקודה הקטנה

ביותר שיכול המחשב לרשום על המסך.

כל נקודה כזאת נקראת באנגלית (PICTURE ELEMENT) PIXEL.

גם אנחנו נקרא לנקודה כזאת פיקסל.

מעבר ממסך למסך

?

כיצד תוכל להציג עכשיו את התוכנית?

לשם כך עליך לחזור למסך הרגיל! כיצד תעשה זאת?

ישנן כמה שיטות. אחת מהן היא שימוש במקשי ה-RUN-STOP

וה-RESTORE (כפי שלמדת ביחידה 1 עמוד 33).

לאחר שלחצת על מקשים אלה, המחשב החזיר אותך למסך הרגיל.

ועכשיו הצג את התוכנית.

הוסף עוד שורה לתוכנית:

20 PLOT 200, 100

וכן שורה כזאת שהמחשב יציג באופן אוטומטי את המסך הגרפי,

בכל פעם שתריץ את התוכנית.

(תשובה 1 בעמוד 62)

הרץ את התוכנית ובדוק את התוצאה.

זכור!

המחשב יכול להציג את התוכנית (LIST) רק במסך הרגיל ואילו

רק במסך הגרפי יכול המחשב להציג פיקסלים (כמו השניים

המופיעים עכשיו).

TEXT

כדי לעבור מהמסך הרגיל למסך הגרפי חייבים לפקוד על

המחשב _____.

כבר אמרנו כי כדי לעבור מהמסך הגרפי למסך הרגיל ישנן כמה

שיטות.

שיטה אחת, שכבר השתמשת בה, היא שימוש במקשים _____

ו- _____.

שיטה נוספת היא שימוש בפקודת TEXT (טֶקְסְט) (הקיצור לפקודת

ה-TEXT הוא T ו-E-SHIFT).


הקש TEXT על המקשים - שום דבר לא נראה במסך הגרפי!

אך מה קורה כאשר תלחץ עכשיו על מקש - RETURN?


המחשב מבצע את הפקודה וחוזר להציג את המסך הרגיל.
 שים לב! פקודת TEXT מחזירה אותך למסך הרגיל מבלי למחוק את
 המסך הרגיל או את המסך הגרפי!

שני המסכים והזכרון

ביחידה 3 פרק א למדת, כי ישנם תאי זכרון קבועים השומרים
 בתוכם את האינפורמציה לגבי הסימנים המודפסים על מסך
 הטלויזיה. כזכור, מספרו של תא הזכרון הראשון הוא 1024.
 במצב הרגיל המחשב משדר לטלויזיה את האינפורמציה הנמצאת
 בתאים אלו.



התוכנית שהעלית מהקסטה לזכרון המחשב גרמה לו, בין השאר,
 להקצות קבוצת תאים נוספת בזכרון. המחשב יכול לשדר גם את
 האינפורמציה הנמצאת בהם אל הטלויזיה.
 פקודת ה-GRAPHICS היא שגורמת למחשב לשדר לטלויזיה את
 האינפורמציה שבתאים אלה:



פקוד על המחשב לעבור למסך הגרפי.

הקש על מקשים שונים!

מדוע המחשב אינו מציג אותם על המסך הגרפי?

פשוט מאד

לחיצה על המקשים משנה את אשר במצא בתאי הזכרון של המסך

הרגיל - ועכשיו המחשב אינו משדר אותם לטלויזיה!

לחץ RETURN.

לאחר שלחצת RETURN מציג המחשב את המסך הרגיל.

עכשיו אתה גם יכול לראות את כל המקשים שלחצת עליהם קודם.

שים לב להודעת ה-SYNTAX ERROR המופיעה על המסך!

מהי הסיבה לכך?

כאשר לחצת RETURN המחשב לא הבין מה בדיוק רצית שהוא יבצע!

זכור!

בכל פעם שתמצא במסך הגרפי ותפקוד על המחשב לבצע פקודה לא

חוקית - המחשב יחזור למסך הרגיל ויודיע לך כי ישנה שגיאה

בפקודה!

פקוד על המחשב לעבור למסך הגרפי.

לחץ על מקש של אחת האותיות. מה יקרה אם תלחץ עכשיו

RETURN? חשוב לפני שתנסה!

פקוד על המחשב לבצע PLOT 100, 50.

מה יתקבל על מסך הטלויזיה לאחר שתפקוד GRAPHICS?

הסבר:

ביצוע פקודת ה-PLOT גרמה למחשב לשנות את אחד מתאי הזכרון

של המסך הגרפי ולכן כאשר פקדת על המחשב לשדר את המסך הגרפי

(GRAPHICS) אל מסך הטלויזיה הופיע הפיקסל השלישי!

עכשיו, כאשר אתה נמצא במסך הגרפי -
פקוד על המחשב לבצע 1000 , 1000 PLOT מבלי שתראה את הפקודה.
לאחר שלחצת RETURN המחשב קלט וביצע את הפקודה ונקודה
רביעית הופיעה על המסך.

CLS – ניקוי המסך הגרפי

פקוד על המחשב CLS.

לאחר שלחצת RETURN - הפיקסלים במחקו מהמסך הגרפי!
CLS גורמת למחשב לנקות את תאי הזכרון של המסך הגרפי, ולכן
הפיקסלים בעלמו מהמסך.
(CLS - קיצור של CLEAR SCREEN - בקה את המסך.)
חזור למסך הרגיל ובה את התוכנית הקיימת מזכרוננו של המחשב.

נסכם:

- לאחר טעינת הקסטה למחשב יש בזכרוננו תאי זכרון מיוחדים המיועדים לשמור את האינפורמציה עבור המסך הגרפי.
- פקודת _____ גורמת למחשב "לשדר" למסך הטלויזיה את האינפורמציה של המסך הגרפי.
- פקודת _____ גורמת למחשב להדליק פיקסל במסך ה_____.
- ישנן 3 שיטות לעבור מהמסך הגרפי למסך הרגיל - מהן?
- פקודת CLS מנקה את ה_____.

פרק ב

קואורדינטות וצירים

נקודת הראשית

כתוב את התוכנית הבאה:


```
5  GRAPHICS : CLS
```

```
10 PLOT 0, 0
```

והרץ אותה.

היכן נמצא הפיקסל, המופיע על המסך ביחס לשוליים?

אתה רואה כי המחשב רשם את הפיקסל בנקודה הנמצאת בפינה השמאלית-תחתונה של המסך:



נקודה זו, הנמצאת בפינה השמאלית-תחתונה של המסך נקראת בשם: **נקודת הראשית** $(0, 0)$. (ראשית - התחלה).

BORDER

רוצה להבליט את הפיקסל הלבן?

פקוד על המחשב לצבוע את השוליים בצבע שחור.

ביחידה 3 למדת כי בעזרת POKE לתא זכרון מתאים תוכל לשנות את צבע השוליים. בעזרת התוכנה שהועברה מהקסטה למחשב שלך לא תזדקק לפקודת ה-POKE כדי לצבוע את השוליים.

פקוד על המחשב: `BORDER 0`

(תוכל לכתוב גם פקודה זאת בקיצור: B ו-SHIFT של 0) כפי שאתה רואה, בעזרת פקודת ה-BORDER ניתן לשלוט על **צבע השוליים**.

אגב, פקודת ה-BORDER שולטת על צבע השוליים הן במסך הגרפי והן במסך הרגיל.

רוצה להשתכנע? פקוד על המחשב לחזור למסך הרגיל ופקוד

BORDER Ø

קואורדינטות של נקודה

הוסף את השורות:

7 BORDER Ø

2Ø PLOT 1ØØ, Ø

והרץ את התוכנית.

? היכן רשם המחשב את הפיקסל הנוסף?

? והיכן ירשום המחשב, לדעתך את הפיקסל הזה:

3Ø PLOT Ø, 1ØØ

לאחר שחשבת - הכנס את השורה לתוכנית והרץ אותה.

? איזו פקודה עליך להוסיף לתוכנית (Ø, 1ØØ) • (Ø, Ø) •

על-מנת שהיא תרשום פיקסל רביעי (?) • (Ø, 1ØØ) •

שישלם את הפיקסלים לצורה של

ריבוע דמיוני: (Ø, Ø) • (1ØØ, Ø) •

הוסף את השורה לתוכנית, הרץ ובדוק בעזרת המחשב אם צדקת.

? איזו שורה עליך להוסיף לתוכנית כך שירשם הפיקסל הנמצא

במרכז הריבוע הדמיוני? בדוק תשובתך על המחשב!

נסכם:

כדי שהמחשב ירשום פיקסל במקום כלשהו על המסך עליך להשתמש

בפקודת PLOT ובשני המספרים המלווים אותה:

(מספר שני) , (מספר ראשון) PLOT

המספר השני מציינ את המספר הראשון מציינ את

מיקום הנקודה _____ מיקום הנקודה _____ (מעל, מימין)


(מעל, מימין) לנקודה הראשית.

לנקודה הראשית.

שני מספרים אלה נקראים בשם: הקואורדינטות של הנקודה.

שתי קואורדינטות כאלו מגדירות פיקסל אחד על המסך.

?



הקואורדינטות (200, 100) מגדירות

את נקודה:

הקואורדינטות (200, 0) מגדירות

את נקודה:

הקואורדינטות (50, 150) מגדירות

את נקודה:

הקואורדינטות (0, 150) מגדירות

את נקודה:

בדוק את תשובותיך על המחשב על-ידי הוספת שורות מתאימות.


צירים ראשיים

משימה

עליך לבנות תוכנית שתצייר קו,

שייצא מן הראשית (0, 0) ימינה

ויגיע עד לנקודה הימנית ביותר.



מחק את התוכנית הקודמת וכתוב תוכנית

שתמלא את המשימה. קרא למשתנה שלך - X,

(העזר ב-GO TO).

הכנס בתוך התוכנית שורה כזאת שתגרום למחשב לצבוע את השוליים

בצבע שחור.

(תשובה 2 בעמוד 62)

כאשר הרצת את התוכנית, שרטט המחשב קו לכן - אך בסיום הוא

חזר באופן אוטומטי למסך הרגיל והודיע לך את הודעת השגיאה

הבאה:

? ILLEGAL QUANTITY ERROR IN 20

חזור ופקוד על המחשב להציג את המסך הגרפי.

הקו הלבן המופיע בתחתית המסך הגרפי בקרא:

ציר אופקי ראשי.

? מה ערכו של המשתנה X כאשר התוכנית הפסיקה לרוץ?

(אינך זוכר איך? פנה ליחידה 2 עמוד 10).

? הקואורדינטות של הפיקסל הימני ביותר על הציר האופקי הראשי

(זה הפיקסל הנמצא בפיינה הימנית-התחתונה של המסך) הן:

$(0, 319)$, $(0, 320)$, $(0, 0)$

_____ כדי לבדוק עצמך מחק את הציר האופקי הראשי (CLS) ופקוד על המחשב לרשום את הפיקסל הימני ביותר.

? הציר האופקי הראשי מורכב מ-_____ $(319, 320)$ פיקסלים.

(תשובה 3 בעמוד 62)

• הקואורדינטות של הפיקסל השמאלי ביותר הן $(0, 0)$, $(0, 319)$

• הקואורדינטות של הפיקסל הימני ביותר הן $(0, 0)$, $(0, 319)$

_____ ועכשיו: חקור את **הציר האנכי הראשי**: (הוא יוצא מן הראשית

כלפי מעלה):

• כתוב תוכנית שתצייר את הציר האנכי הראשי.

• צבע את השוליים בשחור (כדי להבחין בגבול בין המסך לבין השוליים).

• קרא למשתנה בשם Y.

לאחר שהרצת את התוכנית ענה:

? • הציר האנכי הראשי מורכב מ-_____ פיקסלים.

• הקואורדינטות של הפיקסל העליון ביותר עליו הן: _____

_____ בדוק עצמך על המחשב - מחק את המסך ופקוד על המחשב להדפיס

את הפיקסל העליון של הציר האנכי.

משימה

כתוב תוכנית שתשרטט תחילה את הציר האופקי הראשי ומיד בהמשך

את הציר האנכי הראשי.

כאשר המחשב יסיים לשרטט את שני הצירים הראשיים הוא ישאר

במסך הגרפי ולא יקפוץ למסך הרגיל.

קרא למשתנה של הציר האופקי - X ולמשתנה של הציר האנכי - Y.

(תשובה 4 בעמוד 62)


אתגר

שנה את התוכנית כך שהמחשב ישרטט בו-זמנית את שני הצירים הראשיים. כלומר - יתחיל לשרטט באותו זמן את שני הצירים וגם יסיים לשרטט אותם באותו זמן.

(תשובה 5 בעמוד 62)

שנה את התוכנית כך,

שהיא תרשום את הריבוע הבא:




וכל זאת בלולאת FOR-NEXT אחת - כלומר כל הקווים מצטיירים ביחד!
(תשובה 6 בעמוד 63)

אל תמחק את התוכנית ועבור הלאה:

משימה

הוסף עוד שורות לתוכנית כך, שהמחשב ירשום בתוך הריבוע עוד פיקסלים שהמרחק ביניהם הוא 5 בכיוון האופקי ו-3 בכיוון האנכי:



משימה

הוסף עוד שורות לתוכנית כך שהמחשב ירשום את האלכסונים של הריבוע!
זקוק לרמז?

בשעת בנית האלכסון היוצא מן הפינה השמאלית-תחתונה של הריבוע - בכל פעם ש-X גדל בפיקסל גם Y גדל בפיקסל...

משימה

בנה פירמידה מלאה שרוחב בסיסה 100 פיקסלים.



100 פיקסלים

זקוק לרמז? ראה חוברת 3 בעמוד 20.

(תשובה 7 בעמוד 63)

כתוב תוכנית, שתרושום על המסך בקודות אקראיות שוב ושוב ללא הפסק.

(תשובה 8 בעמוד 63)

"פיצוח" תוכנית

עליך "לפצח" (להבין) את עקרון פעולתה של התוכנית הבאה:

```
5  GRAPHICS : CLS
20 PLOT ABS(X), ABS(Y)
30 X=X+1
40 Y=Y+1
50 IF Y=199 THEN Y=-Y
60 IF X=319 THEN X=-X
70 GO TO 20
```

כדי לפצח את התוכנית עליך להבין את פעולת ההוראה:
ABS (אבסולוט) - ABSOLUTE שפירושו: **הערך המוחלט**.

פקוד על המחשב:

```
PRINT ABS(-6)
```

המחשב רשם את המספר _____ (6, -6, -1).

כאשר המחשב מבצע את ההוראה ABS הוא הופך מספר שלילי למספר חיובי.

(אינך שולט במספרים שליליים? פנה לנספח א' בעמוד 53).

מה קורה כאשר הוראת ה-ABS מופעלת על מספר חיובי?

פקוד על המחשב להדפיס את הערך של ABS(10).

אתה רואה כי המחשב השאיר את המספר החיובי (10) כמות שהוא.

זכור!

הוראת ה-ABS הופכת מספרים שליליים לחיוביים. במילים אחרות ובשפת המתמטיקאים אומרים, כי היא מוצאת את הערך המוחלט שלהם.


ועכשיו - האם אתה יכול להבין מדוע המחשב, כשהוא מבצע את

התוכנית האחרונה, אינו חורג מגבולות המסך?

(ראה הסבר בתשובה 9 בעמוד 63)

נסכים:

- המסך הגרפי מחולק לרשת עדינה של נקודות (פיקסלים).
- כדי שהמחשב ירשום נקודה במקום כלשהו על המסך יש לדעת את הקואורדינטות שלה.
- הקואורדינטות של כל פיקסל הן מרחקו מן הציר האנכי האופקי: בדוגמא זו:




- הציר האופקי מורכב מ- _____ פיקסלים.
- הציר האנכי מורכב מ- _____ פיקסלים.
- הקואורדינטות (319, 199) הן של הפיקסל הנמצא בפינה ה- _____ של המסך.
- מכמה פיקסלים מורכב המסך כולו? כלומר, כמה פיקסלים יש בסך הכל במסך?

פרק ג


כותבים במסך הגרפי

משחק: פגע במטרה


עכשיו, לאחר שהכרת את הפיקסלים ואת הקואורדינטות שלהם, תוכל לבנות את המשחק: "פגע במטרה". כללי המשחק הם:



1 המחשב משרטט על המסך הגרפי ריבוע קטן (אורך כל צלע - 15 פיקסלים) במקום אקראי על המסך. לאחר שהריבוע מופיע, עליך לבחש את הקואורדינטות של אחד הפיקסלים הנמצאים בתוך הריבוע.



2 לאחר שתחליט מהן הקואורדינטות של אחד הפיקסלים שלדעתך נמצא בתוך הריבוע, תלחץ על מקש כלשהו - ואז המחשב יעבור למסך הרגיל. שם תכניס ב-INPUT את הקואורדינטות. (בניח שניחשת שפיקסל (250, 120) נמצא בריבוע.



3 לאחר שהמחשב קיבל את הקואורדינטות של הפיקסל, שלדעתך נמצא בריבוע, הוא עובר בחזרה למסך הגרפי ומדליק את הפיקסל על המסך, כבוסף לריבוע שכבר ישנו. בדוגמא שלנו, למשל, הפיקסל "פיספס" - ולא נכנס לריבוע.

לאחר לחיצה על אחד המקשים המחשב חוזר להתחלה ומציג ריבוע חדש במקום אקראי על המסך וכך הלאה...

כותבים "באפלה"

אנחנו מתארים לעצמנו כי המעבר בין שני המסכים אינו בוח. שנה את התוכנית כך שלא תעבור כלל למסך הרגיל ותכניס בצורה "עיוורת" את ערכי הקואורדינטות ב-INPUT.

מדפיסים במסך הגרפי

כמובן שגם לגירסה החדשה יש מגבלה: מאד לא נעים להכניס את הקואורדינטות בצורה עיוורת למחשב. כדי להתגבר על כך יש צורך לדעת להדפיס במסך הגרפי. לשם כך עליך ללמוד את השימוש בפקודה:

PRINT AT – הדפס ב...

הוסף את השורה הבאה:

```
1000 GRAPHICS : CLS : PRINT AT 0, 0; "A"
```

(הקיצור ל-PRINT AT הוא AT?)

הרץ את השורה הזאת!

(אינך זוכר איך? RUN 1000 יעזור לך.)

בפינה השמאלית העליונה של המסך הגרפי מופיעה עכשיו אות A!

הוסף את השורה:

```
1010 PRINT AT 39, 0; "B"
```

והרץ את התוכנית משורה 1000.

אתה רואה כי בעזרת הפקודה **PRINT AT הדפס ב...** תוכל לכתוב

גם במסך הגרפי – בכל מקום שתבחר!

לשם כך מחולק המסך הגרפי בדיוק כמו המסך הרגיל.

ישנן בו 40 עמודות (ממספר 0 ועד 39), ו-25 שורות (ממספר 0

ועד 24).

כלומר, המסך מחולק ל-1000 משבצות.

בעמוד האחרון תמצא את רשת המשבצות של המסך הגרפי.

הוסף עוד שורה כך, שהמחשב יכתוב "C" בפינה השמאלית התחתונה של המסך הגרפי.

ושורה נוספת שתגרום למחשב להדפיס "D" בפינה הימנית התחתונה של המסך הגרפי.

(תשובה 11 בעמוד 64)

נסכם:

המספר הראשון המופיע בפקודת ה-PRINT AT מייצג את מספר ה _____ (שורה/עמודה) של המשבצת שבה אתה מעוניין שהמחשב ידפיס.

המספר השני המופיע בפקודת ה-PRINT AT מייצג את מספר ה _____ (העמודה/השורה) של המשבצת.

מה ידפיס המחשב והיכן, לאחר שתוסיף גם את השורה:

```
1040 PRINT AT 1, 23 ; "ABCD"
```

אתה רואה כי המחשב הדפיס את כל המחרוזת.

מספר העמודה והשורה שבפקודת ה-PRINT AT קבע למחשב את מיקום המשבצת של הסימן הראשון שבמחרוזת. כל שאר הסימנים של המחרוזת מודפסים באופן אוטומטי בהמשך.

מצא את הבג

כתוב במחשב:

```
FOR X=ATO 10 : PRINT X : NEXT
```

ולחץ RETURN.

אם העתקת את השורה הזו בדיוק כפי שהיא כתובה למעלה - המחשב

SYNTAX ERROR

הודיע לך

מדוע?

?

(תשובה 12 בעמוד 64)


חזרים למשחק...

הכנס שורות כאלה כך שהמחשב ידפיס במסך הגרפי "X=" בפינה

השמאלית העליונה כאשר הוא נמצא ב-INPUT של הקואורדינטה

האופקית. לאחר שתכניס את המספר ותלחץ RETURN המחשב ידפיס

את ערך הקואורדינטה בהמשך. לדוגמא: $X=250$. מיד לאחר מכן ידפיס המחשב מתחת לשורה זאת " $Y=$ ", כאשר הוא מחכה שתכניס את הקואורדינטה האנכית. כשתסיים להכניס את הקואורדינטה האנכית הוא ידפיס גם את ערכה. לדוגמא: $Y=120$, ואז גם יופיע הפיקסל המתאים:



(תשובה 13 בעמוד 64)


רוצה משחק קצת קשה יותר?

הוסף שורה שתצבע את השוליים בכחול. כעת קצת קשה יותר לנחש את הפיקסל...

OVER 1

אולי לא הבחנת בכך - אבל במשחק שפיתחת זה עתה יש "בג" קטן - אם הפיקסל שניחשת נופל, במקרה, בדיוק על אחת מצלעות הריבוע - לא תבחין בו, כמוכן - ולא תדע היכן הוא נמצא! רוצה להשתכנע?

הוסף עוד קטע לתוכנית כך, שהמחשב ידפיס, מיד לאחר הציור של הריבוע, גם את הקואורדינטות של הפינה השמאלית-תחתונה של הריבוע:



(תשובה 14 בעמוד 64)

כעת הכנס, ב-INPUT, קואורדינטות של פיקסל שיפול בדיוק על אחת מצלעות הריבוע.

האם הפיקסל שהכנסת הופיע?

קרוב לודאי שלא! וזאת מכיוון שהפיקסל הזה כבר "דלוק".
הכנס בשורה 80 את השינוי הבא: (יכול להיות שאצלך מספר
השורה שונה...)

80 PLOT OVER 1 ; X, Y

(את הפקודה OVER תוכל לכתוב בקיצור: 0 ו-V SHIFT)
הרץ את התוכנית והכנס ב-INPUT את הקואורדינטות של הפיקסל
שבפינה השמאלית התחתונה של הריבוע.

שים לב שהפעם, במקום "להדליק" פיקסל שבפינה השמאלית התחתונה
של הריבוע המחשב "כיבה" אותו.

הרץ פעם נוספת את התוכנית.


נסה להדליק פיקסל בתוך הריבוע עצמו!
האם הפעם הפיקסל נדלק?

כיכר פועלת פקודת ה-1 OVER?

כאשר המחשב מקבל את ההוראה **OVER 1** קורה דבר מעניין:
בכל פעם שהמחשב עומד לרשום פיקסל במצב של 1 OVER בנקודה
כלשהי על המסך הוא מבצע בדיקה:
• אם באותה נקודה הפיקסל כבר "דלוק" הוא "מכבה" אותו.
• אם הפיקסל באותה נקודה היה "כבוי" - המחשב "מדליק"
אותו.

1 OVER כמחק

ראינו כי במצב 1 OVER המחשב מכבה פיקסלים דלוקים.
לכן פקודה זו שימושית במחיקת פיקסלים דלוקים:




משימה 1 – תנועה של פיקסל

כתוב תוכנית אשר "תריץ" פיקסל על
המסך משמאל לימין:
הערה: בחוברת 2 עמ' 28 מדובר על
תנועה (אנימציה).

(תשובה 15 בעמוד 65)

משימה 2 – חיצים



כתוב תוכנית אשר תניע חץ - קו
אופקי קצר המורכב מ-5 פיקסלים -
מצד שמאל לצד ימין של המסך:

זקוק לרמז?

נסה להוסיף פיקסל מימין לחץ ולמחוק פיקסל מן הקצה השמאלי
של החץ - ואז החץ ינוע ימינה!

(תשובה 16 בעמוד 65)

? מה יקרה, אם בתוכנית זו, במקום `OVER 1` תכתוב `OVER 0`?
נסה.

אתה רואה כי במצב `OVER 0` המחשב אינו מוחק פיקסלים. למעשה
המחשב תמיד נמצא במצב `OVER 0`, אלא אם כן פקדת עליו `OVER 1`.

ועוד הערה

את פקודת ה-`OVER` אפשר לכתוב לחוד:

`OVER 1`

ואז היא "תופסת" לגבי כל פקודות ה-`PLOT` הבאות אחריה - או
אפשר להכניסה "פנימה": `50, 100 ; PLOT OVER 1`
ואז היא "תופסת" רק לגבי ה-`PLOT` בו היא כתובה.

יש בג

בתוכנית המניעה חץ לרוחב המסך יש בג - בקצה השמאלי של המסך
בשאר חץ "עומד".

נסה להתגבר על בג זה בעזרת הפקודה הבאה:

INVERSE 1

`INVERSE 1` פועל גם הוא כמחק אבל בצורה שונה:

• לא משנה אם הפיקסל דלוק או מכובה - `INVERSE 1` תמיד

מכבה אותו!


התגבר על הבג שבתוכנית הקודמת בעזרת שימוש ב-`INVERSE 1`

במקום `OVER 1`.

שים לב: קיצור ל-`INVERSE` הוא `IN` ו-`SHIFT` של `V`.

פינת המשחק: קליעה למספרים עשרוניים


1



עם תחילת המשחק מופיע בצד ימין של המסך "עמוד" ועליו ספרות מ-0 ועד 3. בצד שמאל של המסך, בגובה אקראי מופיע "חץ":


מטרת המשחק: להעריך (INPUT) היכן בדיוק יתקע החץ בעמוד לאחר שיעוף (אופקית) ימינה:

2



בניח שהערכת שהחץ, לאחר ש"יעוף" ימינה, יתקע בגובה 1.55. אתה "מכניס" מספר זה ב-INPUT, לוחץ RETURN - ואז החץ מתחיל "לעוף" ימינה...

3



... "ובתקע" בעמוד המספרים. המחשב מדפיס בפינת המסך את ההערכה שלך לעומת המיקום המדויק של החץ על גבי עמוד המספרים. בדוגמא שלנו הערכת 1.55 והחץ היה בגובה 1.34.

הכנס שכלולים

- המחשב מחשב את ההפרש בין המיקום המדויק לבין ההערכה (ב-ABS).
 - המחשב מאפשר 10 סיבובים ומדפיס בסוף את ממוצע ההפרשים.
 - הפוך את המשחק למשחק של שניים - אתה נגד חברך.
- אנו לא הבאנו למשימה זאת פתרון - בסה להתמודד בסבלנות עם אתגר זה.

פרק ד

הצבעים נכנסים למסך הגרפי

עד כה הופיעו הפיקסלים והסימנים שהודפסו במסך הגרפי בצבע לבן על רקע כחול.

בפרק זה תלמד כיצד ניתן לשנות את הצבעים הללו.

צביעת הרקע ("נייר") עליו כותב המחשב:

עכשיו נפקוד על המחשב לכתוב על רקע צבעוני.

לשם כך עליך להשתמש בפקודה: **PAPER (פיפּר)** שפירושה: **הרקע - ה"נייר"** עליו יכתוב המחשב.

רשום את שתי השורות הבאות:

```
10 GRAPHICS : CLS
```

```
20 PRINT AT 10, 5 ; PAPER 0 ; "ABC"
```

(גם את הוראת ה-PAPER תוכל לכתוב בקיצור המקובל:

P ו-A SHIFT.)

הרץ את התוכנית.

המחשב כתב את האותיות ABC בצבע לבן על רקע שחור.

(לרקע אנחנו קוראים גם "הנייר" עליו כותב המחשב.)

הערה את שורה 20 ניתן לכתוב גם כך:

```
20 PRINT PAPER 0 ; AT 10, 5 ; "ABC"
```

נסה!

הוסף שורה לתוכנית כך שהמחשב יכתוב את שמך במסך הגרפי על

נייר (רקע) ירוק.

באיזה צבע יופיעו הרווחים בפקודה הבאה:

```
40 PRINT PAPER 12 ; AT 15, 20 ; "X Y"
```

↑

רווחים

נסה וראה!

אתה רואה כי המחשב כותב את הרווחים בצבע הנייר המופיע בתוך הפקודה.

? מה תראה על המסך לאחר שתפקוד:

50 PRINT AT 30, 24 ; PAPER 1 ; "ABCDE"

נסה והיווכח. האם אתה מופתע?

המחשב כותב באופן רגיל את האותיות והסימנים בצבע לבן. ולכן, כאשר הרקע גם הוא לבן - לא תראה כלום! (ממש כתב סתרים)

כיצד ניתן בכל זאת לכתוב על רקע לבן? לשם כך עליך לכתוב את האותיות בצבע כהה - וגם זאת ניתן לעשות!

כתיבה בצבע

כדי לכתוב במסך הגרפי באחד הצבעים עליך להשתמש בהוראת **INK (אינק)** שפירושו **"דיו"**, כלומר - הצבע בעזרתו כותב המחשב. שנה את שורה 50 ל:

50 PRINT AT 30, 24; PAPER 1; INK 0; "ABCDE"

לאחר שתריץ את התוכנית - האותיות שהסתתרו קודם על הנייר הלכך נכתבו בצבע שחור!

זכור!

במסך הגרפי תוכל לפקוד על המחשב לכתוב ב"דיו" (INK) צבעונית על "נייר" (PAPER) צבעוני.

הוסף שורה אשר בה תתן למחשב לכתוב במסך הגרפי משפט בדיו צהובה על נייר סגול.

אם אתה מעוניין שבכל פקודות ה-PRINT AT שתתן למחשב הוא יכתוב על נייר ירוק בדיו צהובה, למשל, אינך צריך לחזור ולכתוב כל פעם 5 PAPER ו-7 INK בתוך כל פקודת PRINT. אתה יכול לתת למחשב שתי פקודות מקדימות, אשר בעקבותיהן הוא תמיד ירשום על נייר ירוק בדיו צהובה:

הוסף את השורות:

70 PAPER 5 : INK 7

80 PRINT AT 15, 0 ; "□□ XY □□"

הוסף עוד שורת PRINT AT ותראה כי המחשב ממשיך לכתוב בדיו צהובה על נייר ירוק.

? מה יקרה אם תוסיף עכשיו את השורה:

100 CLS

הוסף את השורה והרץ את התוכנית!

הסבר

כאשר המחשב מנקה את המסך (CLS) הוא צובע את כל הפיקסלים בצבע הנייר (הרקע). כרגיל צבע הנייר (הרקע) הוא כחול ולכן CLS גרמה לכל הפיקסלים להופיע בצבע כחול. עכשיו, כאשר המחשב קיבל הוראה כללית לכך שצבע הרקע יהיה ירוק - המחשב צבע את כל הפיקסלים בצבע זה כאשר הוא ביצע CLS!

? מה יקרה אם תמחק את שורה 70 מהתוכנית ותריץ אותה?

הכנס בשורה 10 הוראה כזאת שתגרום למחשב לצבוע את כל המסך בצבע סגול (מחק את השורה האחרונה בתוכנית).

(תשובה 17 בעמוד 65)

? מה יקרה אם תכניס לתוך שורה 30 את ההוראה INVERSE 1 כך:

30 PRINT AT 10, 17 ; INVERSE 1 ; "34"

מסקנה: כאשר אתה מכניס את פקודת INVERSE 1 לתוך פקודת PRINT AT, המחשב מחליף את צבעי הדיו והנייר זה בזה.

דיסקו

זוכר את משימת הדיסקו שביצעת ביחידה 3, עמוד 8?
עכשיו חזור וכתוב תוכנית דיסקו במסך הגרפי.

(תשובה 18 בעמוד 65)

משימות קודמות

עכשיו, כאשר אתה שולט בפקודות: PRINT AT, PAPER ו-INK, תוכי למעשה לחזור ולבצע במסך הגרפי את כל המשימות שביצעת בשלוש החוברות הקודמות - החל מ"אי-ציק עושה התעמלות" וכלה ב"מחפשים את המטמון".

אנו ממליצים כי תעשה חלק מהמשימות הקודמות, ובמיוחד את משימת הפירמידה ביחידה 3, עמוד 20, ואת ריבועי-הצבע בעמוד 34.

מחק את התוכנית הקיימת ועבור הלאה...

צביעת פיקסלים

האם תוכל "לנחש" מה יתקבל לאחר שתריץ את התוכנית הבאה:

```
10 GRAPHICS : CLS
20 FOR X=0 TO 199
30 PLOT PAPER 4 ; INK 0 ; X, X
40 NEXT X
```

הרץ את התוכנית.

מה קרה כאן? מדוע קיבלת לאורך האלכסון ריבועים?

כדי להבין את המתרחש הוסף את השורות הבאות:

```
50 GRAPHICS : CLS
60 PLOT PAPER 5 ; INK 0 ; 123, 90
```

הרץ את התוכנית משורה 50.

הוראת ה-PAPER בשורה 60 בתוך פקודת ה-PLOT גרמה למחשב

לצבוע ריבוע קטן בצבע הנייר - במקרה שלנו ירוק.


בתוך ריבוע זה מופיע הפיקסל בצבע שחור - צבע הדיו (INK) שבפקודה.

פקוד על המחשב לצבוע עוד פיקסל בצבע שחור בתוך הריבוע הירוק כך שלא יגע בפיקסל הקודם.

(תשובה 19 בעמוד 65)

פיקסלים ומשבצות

לפניך הגדלה של המשבצת הירוקה שמופיעה על המסך:



אתה רואה כי המשבצת הירוקה בנויה מ-64 ריבועים קטנים (8×8) . כל ריבוע קטן כזה מייצג פיקסל. כלומר - כל משבצת בנויה מ-64 פיקסלים.

בתוך המשבצת הירוקה שהגדלנו אתה רואה את הפיקסל (123, 90) שביקשנו מהמחשב לרשום בצבע שחור.

העזר במשבצת המוגדלת והוסף עוד שורה שתגרום למחשב להדליק את הפיקסלים המסומנים בספרות 1, 2 ו-3 - בצבע שחור.

(תשובה 20 בעמוד 65)

האם המשבצת הירוקה מזכירה לך משהו? ?


ובכן, המשבצת הירוקה היא אחת מהמשבצות שבהם מדפיס המחשב אותיות, מספרים וסימנים, כאשר הוא מבצע את הפקודה PRINT AT.

מה מספר העמודה ומה מספר השורה של המשבצת הירוקה? לשם כך העזר ברשת המופיעה בסוף החוברת.

שים לב!

בחלק התחתון של הרשת שבסוף החוברת ובצד ימין שלה מופיעים זוגות מספרים.


מספרים אלה מציינים את הקואורדינטות של הפיקסלים הקיצוניים בכל משבצת. למשל, המשבצת הירוקה שלנו:



? מה מספר השורה ומה מספר העמודה של המשבצת אשר מכילה את הפיקסל (35, 60), למשל?
(תשובה 21 בעמוד 66)
על המסך שלך עדיין מופיעה המשבצת הירוקה.
השלם את השורה הבאה:

PRINT PAPER 4; INK Ø; AT ? , ? ; "A"

והכנס אותה לתוכנית על-מנת לקבל על המסך את האות "A" - 2
משבצות משמאל למשבצת הירוקה.



פקוד על המחשב לצבוע משבצת משמאל לאות "A" בצבע צהוב וזאת בעזרת פקודת PLOT!

הערה: הקפד שלא תופיע נקודה בתוך המשבצת הצהובה!
עליך לזכור, כי כאשר מופיעה הוראת PAPER בתוך פקודת ה-PLOT, המחשב צובע בצבע הנייר (PAPER) את כל המשבצת אשר בתוכה נמצא הפיקסל.
(תשובה 22 בעמוד 66)

? ועכשיו הסבר לעצמך מדוע מתקבל אלכסון של משבצות סגולות עם קו שחור כאשר מריצים את כל התוכנית.

תוצאה מפתיעה

? מה יקרה אם תפקוד על המחשב להדליק בתוך המשבצת הירוקה פיקסל בוסף לבן.

(שים לב! אינך צריך לתת הוראת INK 1 - ראית כבר כי המחשב מדליק באופן אוטומטי פיקסלים בצבע לבן - אלא אם כן פקדת עליו אחרת.)

הוסף שורה שתבצע זאת. כמו כן הכנס שורת השהייה לפני שהמחשב יבצע שורה זאת. (הכנס השהייה של 1000 "סיבובים").
קרה דבר מעניין!

כאשר המחשב ביצע את הפקודה האחרונה - כל הפיקסלים שכבר הופיעו קודם בצבע דיו שחור הפכו ללבנים!

זכור!

בתוך משבצת כלשהי המחשב יכול לרשום רק בצבע דיו אחד! הוראת הצבע האחרונה היא הקובעת את צבע הדיו של כל הפיקסלים במשבצת

עכשיו הוסף פקודת PLOT בסוף התוכנית כך, שה-A השחור המופיע במשבצת הסגולה - יעלם.

רמז: האות A בנויה מפיקסלים שחורים...

(תשובה 23 בעמוד 66)

פצח את התוכנית

? מה תבצע התוכנית הבאה:

```
10 GRAPHICS : CLS
50 FOR Y=4 TO 199 STEP 8
60 FOR X=4 TO 319 STEP 8
70 PLOT PAPER K ; X, Y
80 K=K+1 : IF K=15 THEN K=0
90 NEXT X
100 NEXT Y
```

לאחר שחשבת - העתק והרץ.

? באיזה צבע מופיעות הנקודות שבתוך המשבצות? מדוע?

(תשובה 24 בעמוד 66)

INK 17

התבונן במסך. ודאי אתה מבחין שבחלק מן המשבצות, ביחוד הבהירות שבהן, אין רואים היטב את הנקודה. שנה את שורה 70:

```
70 PLOT PAPER K ; INK 17 ; X, Y
```

והרץ.

התבונן שוב במסך. במשבצות הכהות מופיעה נקודה בהירה ובמשבצות הבהירות - נקודה כהה.

נסכם:

INK 17 פירושו: הדפס בצבע ניגודי לצבע הנייר!

? איך "תיפטר" מן הנקודות שבתוך המשבצות?

(תשובה 25 בעמוד 66)

נסה לארוג שטיחים בדוגמאות אחרות.

משימה

נסה לקבל אותה תוצאה ללא שימוש בפקודת PLOT!

רמז: העזר ב-PRINT AT.... (תשובה 26 בעמוד 66)

חידה

מה תקבל על המסך אם תריץ את התוכנית הבאה במשך כשעה:

```
10 GRAPHICS : CLS
20 PLOT INK (RND (1)*16) ; INT (RND (1)*320),
   INT (RND (1)*200)
30 GO TO 20
```

נסה...

ולסיום

בפרק זה למדת כיצד לשלב צבע במסך הגרפי.
עליך לזכור את מגבלות הצביעה במסך הגרפי:

- בעזרת הפקודה _____ (PAPER ,INK) אתה יכול לצבוע את הפיקסלים ב-16 צבעי הקומודור.

- בעזרת הפקודה _____ (PAPER ,INK) ניתן לצבוע את הרקע ("הנייר"). הנייר הקטן ביותר הוא בגודל משבצת אחת.

עליך לזכור את מגבלות הצביעה של הפיקסלים והנייר:

- לא תוכל לרשום שני פיקסלים בעלי צבע שונה באותה משבצת
- באותה משבצת לא יופיעו שני צבעי נייר (או יותר).
- כאשר המחשב מבצע את פקודת ה-PLOT הוא מדליק פיקסל בצבע _____ (PAPER ,INK).

כאשר המחשב מבצע 1 PLOT OVER:

?

- אם הפיקסל "דלוק" (בצבע הדיו) המחשב הופך אותו לצבע ה _____ של המשבצת.
- אם הפיקסל "מכובה" (מופיע בצבע הנייר) המחשב הופך אותו לצבע ה _____.
- 1 PLOT INVERSE הופך תמיד את הפיקסל לצבע ה _____ של המשבצת.
- מיד לאחר ביצוע פקודת GRAPHICS המחשב כותב בדיו _____ על נייר _____ ב _____ OVER וב _____ INVERSE.

פרק ה

עפרון וידאו עדין

משימה

זוכר את עיפרון-הוידאו מיחידה 2?
עליך לבנות עיפרון-וידאו "עדין", המצייר קוים דקים (בעזרת PLOT) במקום קוים עבים (משבצות).
יש בידך כל הכלים. (אם תתקשה תוכל להעזר ביחידה 2 עמוד 45).
(תשובה 27 בעמוד 66)

אותה גברת – עם אדרת קצרה יותר...

כזכור, סוד פעולתו של עיפרון-הוידאו נעוץ בשורה מהסוג:

```
GET A$ : IF A$= "1" THEN X=X+1
```

ביתן לכתוב שורה כזאת כך:

```
GET A$ : X=X-(A$= "1")
```

מבחינת המחשב, 2 השורות זהות לחלוטין!

מה מבין המחשב בביטוי: (A\$ = "1")

המחשב בודק אם מה שכתוב בתוך הסוגריים הוא "נכון", כלומר

אם מתקיים: "1" = A\$, כלומר אם לחצת על המקש המזיז את

המסמן ימינה.

- אם התנאי מתקיים (כלומר מה שכתוב בתוך הסוגריים נכון) - כל הסוגריים מקבלים את הערך: 1-
- אם התנאי אינו מתקיים - כלומר לא לחצת על המקש המביע את המסמן ימינה - הביטוי שבתוך הסוגריים מקבל את הערך: 0

הערה לפני הסוגריים שמנו סימן מינוס (-) כי כשהתנאי מתק הסוגריים שווים: -1. ואז: $-(-1)=+1$.
(אינך מבין? פנה ללמוד את בספח א': מספרים שליליים בעמוד 53.)

נראה אם הבנת את העניין:

למה שווה כל אחד מן הביטויים הבאים? **?**

- | | |
|----------------|-----------------|
| א. $(2=3)$ | ה. $(-2 > 0)$ |
| ב. $(-2 < -3)$ | ו. $-(3 > 1)$ |
| ג. $(-2 > -3)$ | ז. $-(-3 < -1)$ |
| ד. $(2 < > 3)$ | |

בדוק תשובותיך על המחשב בעזרת PRINT, למשל:

PRINT (2=3)

כאמור, ניתן בעזרת כלי זה לקצר את תוכנית עיפרון-הוידאו מה תפקיד השורה הבאה: **?**

$X=X-(A\$ = "1") + (A\$ = "1")$


שנה את תוכנית עיפרון-הוידאו וקצר אותה תוך שימוש בביטו סוגריים.

(תשובה 28 בעמוד 7)

נסכם:

"ביטוי סוגריים" שלמדת עליו זה עתה בקרא גם **ביטוי לוגי**.

הביטוי שווה ל- $(-1, 0)$ אם התנאי שבתוך הסוגריים מתקיים (נכון), ושווה ל- $(-1, 0)$ אם מה שכתוב בסוגריים אינו נכון. לעיתים קרובות ביטוי כזה מחליף בהצל שורת IF... THEN ומקצר את התוכנית.



משימה

עליך לשכלל את התוכנית כך, שתוכל לצייר גם צורות שאינן מחוברות ביניהן:

(תשובה 29 בעמוד 7)



אל תמחק את התוכנית!

שכפול צורות של עפרון – וידאו

ביחידה 2 עסקנו בשכפול צורות. מה דעתך על התוכנית הבאה:

1

אתה מצייר בעזרת עיפרון-הוידאו את הצורה הבאה על המסך:
(או כל צורה אחרת)



לאחר לחיצה על אחד המקשים
שתיכנתת, המחשב משכפל בדיוק את
אותה צורה על פני כל המסך!

2

כדי לפתח את התוכנית עליך להשתמש בשני מערכים (DIM) אשר
בעזרתם יוכל המחשב לזכור באילו פיקסלים (X, Y) עבר המחשב.
כשתפקוד עליו - הוא יוכל לשחזר את הצורה.
(אינך זוכר מהו מערך? - פנה ליחידה 3 עמוד 49).

ושוב בידך כל הכלים הדרושים - בסה לבנות את התוכנית בכוחות
עצמך!
לא הצלחת? בעבוד ביחד שלב אחרי שלב:

שלב 1

בשלב זה עליך להוציא מן התוכנית את כל השורות העוסקות
ב"מחק" של הפיקסלים. בשלב מאוחר יותר נחזיר אותן לתוכנית.
לכן אל תמחק אותן אלא "נטרל" אותן בעזרת REM.
(אינך זוכר? ראה חוברת 3 עמוד 32).

שלב 2

בשלב זה עליך לפקוד על המחשב "לזכור" את כל הקואורדינטות של הפיקסלים שהוא מצייר. לשם כך עליך להשתמש במערכים.

נסה להשלים את התוכנית בכוחות עצמך.

הערה

הבג הנפוץ ביותר בשלב זה הוא, שלמרות שאינך מזיז את "העיפרון" כלל - המערכים הולכים ומתמלאים! אנו ממליצים על שורה כמו:

```
110 GET A$ : IF A$ = "" THEN 110
```

(תשובה 30 בעמוד 67)

שלב 3

בלחיצת מקש - המחשב מוחק את המסך ומתחיל לשכפל את הצורה שציירת - שוב ושוב על פני כל המסך! (ראה עמוד 35). את שלב 3 אפשר לחלק לשלבי-משנה:


- המחשב מוחק את הציור - ומצייר אותו מחדש באותו מקום.
- המחשב מוחק את הציור - ומצייר אותו מחדש למטה משמאל.
- המחשב מוחק את הציור - ומצייר למטה שורה של ציורים כאלה.

ולבסוף:

- לאחר שהמחשב מוחק את המסך - הוא משכפל כמה שורות של ציורים כמו שציירת בעיפרון-הוידאו.


(תשובה 31 בעמוד 67)

שכלול מדהים

בעזרת שינויים קלים בתוכנית תוכל לפקוד על המחשב להקטין או להגדיל את הציור שלך! (במקום לשכפל אותו) בסה! 

(תשובה 32 בעמוד 67)

עפרון קלידיאוסקופי

שנה את התוכנית כך, שבזמן הציור שלך - המחשב יצייר את 

תמונת הראי:



תאר לעצמך שבאמצע המסך מונח ראי דמיוני בצורה אנכית:

כשאתה מושך קו ימינה - המחשב מצייר גם קו שמאלה.

כשאתה "עולה" למעלה - גם המחשב עולה למעלה.

הוסף גם ראי אופקי - אתה מצייר קו אחד - והמחשב מוסיף עוד

שלושה! (2 בראי האופקי ועוד אחד בראי האנכי, הקיים)

(תשובה 33 בעמוד 68)

מעוניין בעוד כמה שכלולים?

- פקוד על המחשב להדפיס בפינת המסך, בכל רגע ורגע, בכמה פיקסלים כבר השתמשת בציור.
- פקוד על המחשב לא להמשיך ולנסות "לטעון" את המערכים אם מספר הפיקסלים עולה על גודל המערכים.
- החזר את השורות העוסקות ב"מחק" (אילו שניטרלת בעזרת REM) - פתור את הבגים הנוצרים כתוצאה מכך!

משימה לסיום הפרק

לסיום הפרק נציב בפניך אתגר מחשבתי מסוג חדש.
הנה תוכנית של משחק שקראנו לו: "מגלה מוקשים".

```
30 X=4 : Y=4
40 GRAPHICS : CLS
50 A=INT (RND (1)*320)
60 B=INT (RND (1)*200)
70 GET A$
80 X=X-(A$ = "→") + (A$ = "←")
90 Y=Y-(A$ = "↑") + (A$ = "↓")
100 PLOT PAPER N; INK 17; X, Y
150 N=(ABS(X-A)+ABS(Y-B))/35
200 IF A$ = "+" THEN 300
210 GOSUB 500
220 GO TO 70
300 CLS
320 PLOT INK 17; A,B : PLOT X,Y :PLOT X+1, Y :
      PLOT X+1, Y+1: PLOT X, Y+1
400 END
500 V=54296 : W=54276 : R=54277 : H=54273
510 POKE V, 15 : POKE W, 17 : POKE R, 15 :
      POKE H, 60-N*4
520 FOR T=1 TO 20 : NEXT
530 POKE W, 0 : POKE R, 0
540 RETURN
```

לא בתאר לך את שלבי המשחק ואת כלליו. משימתך היא לגלות אור
בעצמך!

(אם בכל זאת התקשית - ראה תשובה 34 בעמוד 68)

פרק ו

מתיחת קוים על המסך

פקודת DRAW

עד כה שירטטת קוים עדינים בעזרת פקודת PLOT - באיטיות די-רבה.

המחשב מאפשר לך למתוח קוים על המסך במהירות ובקלות בעזרת פקודה מיוחדת.

■ בקה את זכרון המחשב מכל תוכנית קודמת.

■ העתק את התוכנית הבאה למחשב:


```
5  GRAPHICS : CLS
10 PLOT 100, 50
20 FOR T=1 TO 500 : NEXT
30 DRAW 120, 80
```

הערה: וכרגיל - קיצור של DRAW הוא D ו-SHIFT של R.

פקודת **DRAW (דרו)** - אומרת למחשב לשרטט קו.

■ הרץ את התוכנית הקצרה.

אתה רואה כי המחשב מצייר את הפיקסל (100, 50), ממתיך כשניה ואח"כ מותח קו מן הפיקסל ואילך!



שים לב!

- הפיקסל האחרון שהמחשב הגיע אליו לפני תחילת ביצוע פקודת ה-DRAW משמש כנקודת התחלה לשרטוט הקו. (במקרה שלנו זה הפיקסל 50, 100).
- המספר הראשון בפקודת ה-DRAW הוא המרחק האופקי של נקודת הסיום מנקודת ההתחלה (במקרה שלנו 120).
- המספר השני הוא המרחק האנכי של נקודת הסיום מנקודת ההתחלה (במקרה שלנו 80).

השלם את התוכנית הבאה:

```
5  GRAPHICS : CLS
10 PLOT 40, 50
20 FOR T=1 TO 500 : NEXT
30 DRAW ? , ?
```

כך, שהמחשב ימתח קו מהפיקסל (40, 50) עד לפינה הימנית-עליונה של המסך.

(תשובה 35 בעמוד 68)

הוסף עוד שורה לתוכנית כך, שהמחשב ימתח קו מאותו פיקסל (40, 50) אל הפינה השמאלית-עליונה של המסך:

- רמזים: המחשב מתחיל את ה-DRAW מן הפיקסל האחרון שהוא הגיע אליו.
- $0-40 = -40$

(לא ברור? - ראה בספח מספרים שליליים).

(תשובה 36 בעמוד 68)

הוסף עוד שורות לתוכנית כך שימתחו עוד 2 קוים מהפיקסל (40, 50) אל שתי הפינות הנותרות של המסך.

(תשובה 37 בעמוד 68)

אגב:

אם תכתוב NEW, ואח"כ DRAW 100, 100 ?

מה תהיה נקודת ההתחלה של הקו?


נסה.

מסקנה


- המחשב מתחיל את ה-DRAW מן הפיקסל האחרון שהוא הגיע אליו.
- אם הוא לא הגיע בתוכנית לשום פיקסל לפני ה-DRAW - יתחיל המחשב את מתיחת הקו מן הראשית!

רשת קורי העכביש

כתוב תוכנית שתבצע את הפעולה הבאה:
כתבת RUN - והתוכנית מציירת על
המסך פיקסל ומותחת קוים ממנו
לארבע פינות המסך:



1



2

לחצת על מקש המסמן ימינה - הפיקסל
"קפץ" 5 פיקסלים ימינה והמחשב
צייר מיד 4 קוים ל-4 פינות המסך:


וכך הלאה: בעזרת מקשי עיפרון-הוידאו המחשב מצייר פיקסלים
(בקפיצות של 5) ומותח קוים לפינות המסך. מתקבלת רשת באה
של "קורי-עכביש".

(תשובה 38 בעמוד 68)

סיימת להשתעשע עם התוכנית? מחק אותה ועבור הלאה:

אריגת שטיחים

כתוב תוכנית שתצייר על כל המסך קוים אנכיים במרחק של 2



פיקסלים זה מזה:

רמז: העזר ב-FOR ...NEXT ...STEP

אגב: הקוים מופיעים בצבעים שונים

רק בגלל כושר ההפרדה הנמוך של

מסך הטלוויזיה.

(תשובה 39 בעמוד 69)

הוסף עוד שורות לתוכנית כך, שהמסך יתמלא גם בקוים אופקיים


(תשובה 40 בעמוד 69) באותו מרווח בין קו לקו.

עד עכשיו התוכנית בראית משעממת לחלוטין...

הוסף שינוי קל לתוכנית כך, שאחרי שהמחשב מסיים לצייר את הקווים האנכיים והאופקיים - ה-STEP גדל ב-1 והתוכנית חוזרת על עצמה - וכל זאת ב-1-OVER.

מתקבלת תוצאה מרהיבה ביופיה של שטיחים הנארגים זה על זה ללא הפסק - הנח לתוכנית לעבוד עד שהמסך מפסיק להשתנות! (תשובה 41 בעמוד 69)

- שחק עם ה-STEP ותקבל צורות בוספות.
- צבע את השטיחים כרצונך.




שטיח פרסי

בחר נקודה במרכז המסך ומתח ממנה קווים אל פיקסלים הנמצאים על הגבול העליון של המסך בקפיצות (STEP) של 3:

1

(תשובה 42 בעמוד 69)



(תשובה 43 בעמוד 69)

המשך ומתח קווים מהפיקסל שבמרכז אל פיקסלים הנמצאים לאורך כל גבולות המסך (באותו STEP).

2

שחק עם השטיח:

- חזור על "אריגת השטיח" בצבעים שונים.
 - בצע את כל התוכנית עם 1-OVER.
 - עבור שוב על השטיח עם STEP הולך וגדל.
- מה תבצע התוכנית הבאה: ?

```
5 GRAPHICS : CLS
```

```
7 FOR N=1 TO 19
```

```
10 DRAW 10, 0
```

```
20 DRAW 0, 10
```

```
30 NEXT N
```

הרץ ותיווכח.

משימה

כתוב תוכנית, ללא אף פקודת PLOT, שתצייר מלבן בדיוק על גבולות המסך (המלבן הגדול ביותר שניתן לצייר).


(תשובה 44 בעמוד 70)

שתול, לתוך פקודות ה-DRAW, פקודות PAPER כך, שהמסגרת תהיה ירוקה ועבה.


מלבנים מתכווצים

שנה את התוכנית הקודמת כך, שאחרי ציור המלבן הראשון, התוכנית מציירת עוד מלבן, קצת יותר קטן, בצבע אחר, הנכנס בדיוק לתוך המלבן הקודם. כך הלאה עד שכל המסך מתמלא במלבנים צבעוניים - זה בתוך זה.


(תשובה 45 בעמוד 70)



(תשובה 46 בעמוד 70)



(תשובה 47 בעמוד 70)



"פצח את הציור"

כתוב תוכנית אשר תצייר את הדוגמא הבאה במסך הגרפי:

משימות נוספות

מלבנים

כתוב תוכנית שתשרטט מלבן צבעוני:

מסגרות

שנה את התוכנית הקודמת כדי לקבל מסגרת:

יש כמה דרכים לפתרון!

דרך אחת תראה בתשובה 48 בעמוד 70.

מלבנים מתחלפים



המחשב מצייר ריבוע קטן מלא:

1

2

3

?



המחשב מצייר על הריבוע הראשון, ריבוע בוסף מלא,

גדול יותר, ב-1 OVER:

המחשב הולך ומוסיף, ב-1 OVER,

ריבועים גדולים יותר ויותר על


הצורות הקיימות כבר על המסך.

(תשובה 49 בעמוד 71)

לפני שתצא לשחק בשדה הגולף כדאי שתדע לענות על השאלה הבאה:

התאם בין הפקודות לבין הקווים שהפקודות ציירו על המסך:



1. PLOT 125,85: DRAW 50,50
2. PLOT 125,85: DRAW -50,50
3. PLOT 125,85: DRAW 50,-50
4. PLOT 125,85: DRAW -50,-50



בדוק תשובותיך על המחשב!

פינת המשחק: בשדה הגולף

הידע שצברת עד כה מאפשר לך לפתח משחק קטן המזכיר את משחק הגולף:



בתחילת המשחק המחשב מצייר על המסך,

במקומות אקראיים, 5 ריבועים קטנים,

ובתוך כל ריבוע מספר סידורי.

בשלב ראשון עליך למתוח מפיקסל (0,0)

קו שקצהו יהיה בתוך ריבוע מספר 1.

המחשב מחכה ממך ל-INPUT של A ושל


B - עליך להעריך את גודלם על-מנת

שהקו שהמחשב ימתח יסתיים בתוך ריבוע 1.

1


2

בדוגמא שמשמאל - בסיון א' לא הצליח
ורק בסיון ב' הצליח להכנס לתוך
הריבוע.



כעת יש להמשיך ולנסות להכנס לתוך ריבוע 2.

מטרת המשחק: "לבקר" בכל הבורות (הריבועים) במספר קטן ככל האפשר של נסיונות! (המחשב סופר ומדפיס בפינת המסך את מספר הנסיונות עד כה.)



התפוצצויות בחלל

כתוב תוכנית שהתוצאה
שלה תהיה הציור הבא:

כלומר, המחשב מצייר בקודה במרכז המסך ומושך ממנה קוים באורך אקראי לכיוון אקראי, שוב ושוב ללא הפסק (OVER 1).
רמז: פתור את הבעיה הבאה:

- כתוב תוכנית שמגרילה באקראי מספרים בין -50 לבין $+50$.
(תשובה 51 בעמוד 71)

המחשב המקשקש

כתוב תוכנית אשר בה המחשב מצייר (DRAW) קו באורך אקראי ובכיוון אקראי - ועוד קו - ועוד קו. כל קו מתחיל, כמוכח, במקום בו הקו הקודם הסתיים. התוצאה היא קשקוש מעניין על המסך. אם המחשב יהיה ב-OVER 1 "הציור" יהיה מעניין יותר.

הערה

הבעיה העיקרית בתוכנית זו היא איך לודא שהמחשב לא יחרוג מן המסך - ובכל זאת ימלא את כולו! (תשובה 52 בעמוד 72)

שאלה לסיכום הפרק.

כאשר בפקודת ה-DRAW A, B:

A. הוא מספר שלילי אז תמיד ימתח המחשב קו הנמצא _____

(משמאל, מימין, למעלה, למטה) מנקודת ההתחלה.


וכאשר B הוא מספר שלילי אז תמיד ימתח המחשב קו הנמצא _____

(משמאל, מימין, למעלה, למטה) מנקודת ההתחלה.

פרק ז

משפט פיתגורס

בפרק זה תלמד כיצד ניתן לשלב בתוכניות הגרפיות שלך אחד מהכלים המתמטיים השימושיים ביותר שהיה ידוע כבר בעת העתיקה. לפניך משולש ישר-זווית:



(זהו משולש שאחת מהזוויות שלו שווה ל-90 מעלות). שתיים מהצלעות **מאונכות** זו לזו.

הצלעות המאונכות (ניצבות) זו לזו הן צלע (A, B, C) וצלע (A, B, C) .

קח סרגל, ומדוד (ב-מ"מ) את אורך כל אחד מהניצבים: (הצלעות המאונכות)

A = _____

B = _____

הצלע השלישית, C, הארוכה מכולן, בקראת בשם **יתר** והיא נמצאת מול הזווית הישרה. אורכה (ב-מ"מ) במשולש שלנו הוא: C = _____ (מדוד בעזרת סרגל)

משפט פיתגורס אומר כי במשולש ישר זווית תמיד קיים הקשר הבא בין הצלעות:

$$c * c = a * a + b * b$$


רוצה להשתכנע כי הדבר נכון?

פקוד על המחשב לחשב ולרשום על המסך את התוצאה של $A*A+B*B$ כפי שמדדת אותם במשולש שלנו.

כעת פקוד עליו לרשום את $C*C$.

אם דייקת במדידותיך תראה כי התקבלו תוצאות מאוד דומות.


במילים אחרות משפט פיתגורס קובע ש:



אם נבנה ריבוע על היתר (ששטחו C^2) הוא יהיה שווה בשטחו לסכום השטחים של ריבועים הבנויים על הניצבים (ששטחם: $A^2 + B^2$).

תרגיל מתמטי

מה אורכו של היתר במשולש בו אורכי הניצבים הם 3 ס"מ ו-4 ס"מ? בסה לפתור זאת ללא המחשב.



(תשובה 53 בעמוד 72)

תרגיל מתמטי

?

חשב, בעזרת המחשב, את אורך היתר, אם ידוע כי אורך ניצב אחד הוא 76 ס"מ ואורך הניצב השני - 34 ס"מ.

(תשובה 54 בעמוד 72)

כיצד פתרת שאלה זו?

ודאי לא היתה לך בעיה לחשב: $6932 = 76^2 + 34^2$, אך כעת עליך למצוא מספר כזה שאם בכפיל אותו בעצמו נקבל 6932! בשפת המתמטיקאים קוראים למספר כזה בשם **שורש ריבועי** של 6932. במחשב יש פקודה מיוחדת המחשבת שורשים כאלה:

(SQUARE ROOT) SQR

פקוד על המחשב: PRINT SQR (6932)

הכפל את המספר שהתקבל בעצמו ותקבל 6932.


לכן המספר שקבלת (83.2586333) הוא **אורך היתר**.

עוד תרגיל מתמטי


?

שטחו של ריבוע הוא 30 ס"מ מרובעים.

מה אורך צלע הריבוע A?



(תשובה 55 בעמוד 72)



"מקל מחליק על קיר"

עליך לכתוב תוכנית המציירת על המסך את כל השלבים של "מקל" באורך L הנשען על קיר ומחליק על הרצפה עד שהוא נופל ומונח על הרצפה:
זקוק לרמזים?:

- אם $C^2 = A^2 + B^2$ אז גם $B^2 = C^2 - A^2$.
- בנה תחילה את התוכנית כך, שהמחשב ישאל אותך מה גודלו של X ב-INPUT, ואז יצייר לך את המקל. אח"כ מחק את שורת ה-INPUT ושנה את X ב-FOR-NEXT בקפיצות של 5 פיקסלים, למשל.


(תשובה 56 בעמוד 72)

שכלולים

- אחרי שהמקל מונח על הרצפה הוא מתחיל להתרומם לאורך הקיר הנגדי.
- המקל מטייל גם על התיקרה!
- בסה לשחק עם 1 OVER ועם צבעים.

פיתגורס מצייר מעגל

בעזרת משפט פיתגורס תוכל גם לצייר מעגלים.



?
בניח כי המעגל הבא הוצג על מסך הטלוויזיה:


- מהן הקואורדינטות של מרכז המעגל?
- מהו הרדיוס של מעגל זה?

(להזכירך: רדיוס המעגל הוא המרחק של הנקודות היושבות על המעגל ממרכז המעגל.)

(תשובה 57 בעמוד 72)

משימה

עליך לצייר על מסך הטלויזיה את המעגל המשורטט למעלה.
וזאת בעזרת משפט פיתגורס.
אם אתה מתקשה (סביר בהחלט...) העזר ברמזים הבאים:



רמז 1 בשלב ראשון עליך להדליק את אחד הפיקסלים הנמצאים על המעגל. מרכז המעגל נמצא בנקודה ידועה (XC, YC) (אנו משתמשים בשמות אלו למשתנים כיוון ש-C היא האות הראשונה של המילה CENTER - מרכז).

הרדיוס הוא R (ואורכו ידוע).

P היא נקודה כלשהי על המעגל. אם בדע את הגדלים של הניצבים A ו-B בדע להדליק את הפיקסל P (ראה ציור).

כתוב תוכנית, אשר בה אתה מכניס ב-INPUT את הגודל A -

והמחשב מדליק את הפיקסל P. (הדלק קודם את מרכז המעגל).

(תשובה 58 בעמוד 72)

רמז 2

במקום INPUT פקוד על המחשב לשנות את A (FOR-NEXT) מ: -R

עד: +R

הערה: אתה מתקשה במספרים שליליים - ראה בספח א.

(תשובה 59 בעמוד 73)

רמז 3


קיבלת חצי מעגל, כדי להשלים

אותו למעגל שלם - עליך להדליק

פיקסל בחצי המעגל התחתון בכל

פעם שאתה מדליק פיקסל בחצי

העליון:



(תשובה 60 בעמוד 73)

משימות מעגליות נוספות

משימה 1

שנה את תוכנית המעגל כך,
שיתקבל עיגול מלא:



רמז: במקום להדליק את הפיקסל בחצי המעגל התחתון - מתח אליו
קו מהפיקסל שבחצי המעגל העליון.

(תשובה 61 בעמוד 73)

משימה 2

על בסיס התוכנית הקודמת כתוב תוכנית שתצייר עיגול מלא קטן,
עליו היא תצייר, ב-1 OVER, עיגול מלא גדול יותר וכן הלאה
(ראה עמוד 44).

(תשובה 62 בעמוד 73)



משימה 3

שיבוי קל בתוכנית הקודמת יגרום למחשב לצייר אליפסות זו על
גבי זו: הכפל את B ב-2 למשל.

(תשובה 63 בעמוד 73)

משימה 4

שנה את התוכנית כך, שיתקבלו מעגלים זה בתוך זה כמו
בציורים:




(תשובה 64 בעמוד 73)

משימה 5

שנה את התוכנית כך, שתצייר מעגל אחד, המצטייר מ-4 כיוונים

בבת אחת:




(תשובה 65 בעמוד 73)

משימה 6

שנה שוב את התוכנית כך, שמכל בקודה על המעגל ימתחו קווים אל

הבקודה הנגדית, שמעבר לקוטר, ואל המרכז, כך שיווצרו

משולשים.



(תשובה 66 בעמוד 73)

המשך לשחק עם התוכנית כיד הדמיון הטובה עליך...

שאלת חזרה לסכום

לפניך רשימת מושגים שלמדת ביחידה זו.
עבור על כל אחד מהם ובדוק אם אתה זוכר אותם.
(תוכל לרענן את זכרוןך בעזרת מספרי העמודים המופיעים
בסוגריים ליד כל מושג.):

(16) ABS	(6) GRAPHICS
(19) PRINT AT	(6) PLOT
(21) OVER 1	(6) PIXEL
(23) INVERSE 1	(7) TEXT
(25) PAPER	(10) CLS
(26) INK	(11) נקודת ראשית
(34) ביטוי לוגי	(11) BORDER
(39) DRAW	(12) קואורדינטות של נקודה
(46) משפט פיתגורס	(13) ציר אופקי ראשי
(47) SQR	(14) ציר אנכי ראשי

לסיום...

ביחידת לימוד זאת עשית הכרה עם תחום הגרפיקה העדינה (HIGH RESOLUTION). לימדנו אותך להשתמש בכלים תכנותיים חדשים, בעזרת הקסטה המיוחדת שפנתה לשם כך. הכלים שרכשת כאן מהווים את הבסיס ההכרחי על מנת להמשיך ולהעמיק את הידע שלך בפיתוח תוכניות גרפיות.

שים לב: בספטמבר 1984 תצא לאור "מחשבת-5" בנושא "גרפיקה מתמטית". ביחידת לימוד זאת תשתמש בכל הכלים שלמדת ביחידה (וכמובן גם בקסטה) ותלמד כלים מתמטיים מתקדמים יותר בתחום הטריגונומטריה, כמו הסינוס והקוסינוס וזאת על מנת לפתח תוכניות גרפיות מדהימות. אינך צריך להכיר מושגים אלה לפני שאתה מתחיל ללמוד. אנחנו נלמד אותך צעד אחר צעד כל מושג מתמטי חדש שנעשה בו שימוש.

נספח א

המספרים השליליים

מספרים חיוביים ושליליים

? מה ידפיס המחשב כאשר הוא יבצע את פעולת החיסור הבאה:

PRINT 10-15

המחשב הדפיס את המספר: _____ (5, -5)

למספר כמו (-5) קוראים בשם "**מינוס 5**". למספרים "הרגילים" שאתה מכיר נוהגים לקרוא בשם **מספרים חיוביים**. ואילו למספרים כמו -5 קוראים **מספרים שליליים**.

? אילו מספרים חיוביים ואילו מספרים שליליים ידפיס המחשב על המסך לאחר שיבצע את התוכנית הבאה:

10 X=5

20 FOR N=1 TO 15


30 PRINT X

40 X=X-1

50 NEXT

על המסך קיבלת עמודה של מספרים. המספר החיובי הגדול ביותר הוא _____.

? ומהו המספר השלילי הקטן ביותר? _____
ניתן להמחיש זאת על-ידי הצגת המספרים החיוביים והשליליים על ציר:



ככל שהמספרים ממוקמים ימינה יותר הם גדולים יותר, וככל

שהמספרים ממוקמים שמאלה יותר הם קטנים יותר.

? איזה מספר גדול יותר 0 או -1? (בדוק זאת במחשב)

פעולות חשבון עם מספרים שליליים:

המחשב מתייחס למספרים השליליים כמספרים לכל דבר, והוא יודע לבצע איתם את כל פעולות החשבון. כיצד הוא עושה זאת? זאת נראה עכשיו...

חיבור מספרים שליליים וחיוכיים

מה תהיה לדעתך תוצאת החיבור הבאה:

$$(-7) + (+8) = \underline{\hspace{2cm}}$$

↑ ↑
מספר מספר
שלילי חיובי

בדוק במחשב!

הערה: הכנסנו את הסוגריים רק לצורך בוחות הקריאה של המספרים. המחשב איבנו זקוק במקרה זה לסוגריים. הורד את הסוגריים וראה אם מתקבלת אותה תוצאה:

$$-7 + +8$$

שים לב כי מופיעים שני סימני + בזה אחר זה. מה היה קורה אם במקום שניים היית כותב סימן + אחד? בסה:

$$-7+8 =$$

אתה רואה כי שני סימני + בזה אחר זה שקולים למעשה לסימן + אחד.

ומה היה קורה אם היית פוקד על המחשב:

$$-7+++8$$

בסה!

חיבור מספרים שליליים

מה תהיה התוצאה אם בחבר שני מספרים שליליים?

$$(-7) + (-8) =$$

בדוק עצמך במחשב!

האם התוצאה תשתנה אם תוריד את הסוגריים? בסה!

האם התוצאה תשתנה אם תכתוב: $-7-8$? בסה! ?

מסקנה: אתה ראית כי תוצאת כל התרגילים הבאים

$$(-7) + (-8)$$

$$-7 + -8$$

$$-7 - 8$$

היא זהה ושווה ל-_____.

שים לב! המחשב מתייחס ל-+ או -+ בדיוק כמו: -.

תרגיל: $______ 3 ______ 6 ______ 11 ______ 2 = 0$

שתול סימני פלוס (+) ומינוס (-) על מנת שתתקבל התוצאה 0.
בדוק עצמך במחשב.

תרגיל: ואותו הדבר לגבי התרגיל הבא:

$______ 5 ______ 7 ______ 9 ______ 6 = -15$
בדוק עצמך במחשב.

חיסור מספרים שליליים חיסור מספר חיובי ושלילי

? מה יתקבל אם תפקוד: $(-7) - (+8) =$
בדוק במחשב!

התוצאה הזאת לא צריכה להפתיע אותך כי:

$(-7) - (+8) = -7 - +8 = -7 - 8 = ______$

ראינו כי זה
שקול ל"-"

? מה יתקבל אם תפקוד:

$6 - (+9) =$

בדוק במחשב!

חיסור שני מספרים שליליים

? מה יתקבל אם תפקוד על המחשב לבצע:

$(-7) - (-8)$

האם תתקבלנה תוצאות שונות אם תפקוד על המחשב לבצע את

התרגילים: $-7 - -8$

$-7 + 8$

מסקנה: המחשב מתייחס לפעולה רצופה של שני סימני (--) כאל +.

מה תהיה תוצאת התרגיל הבא:

$-7 - (-9) + (-10) - (-1) =$

בדוק תשובתך במחשב!

? מה ידפיס המחשב אם תפקוד עליו להדפיס ---1

התוצאה לא צריכה להפתיע אותך כי:

$$\boxed{---1} = + - 1 = -1$$

||
+

ומה יהיה אם יופיעו 4 סימני - ?

$$----1 = ?$$

? הוסף סימני פלוס ומינוס כדי לקבל את התוצאה הבאה:

$$(-2) \quad \underline{\quad} \quad (-6) \quad \underline{\quad} \quad (+4) \quad \underline{\quad} \quad (-1) = +1$$


בדוק עצמך במחשב! 

פעולות כפל עם מספרים שליליים

כפל מספר חיובי בשלילי

? מה תקבל אם תפקוד על המחשב:

$$3 * (-7)$$

בדוק זאת! 

התוצאה לא צריכה להפתיע אותך כי הרי:

$$3 * (-7) = (-7) + (-7) + (-7)$$


? ומה יקרה ללא סוגריים? $-7 * 3$ - בדוק במחשב!

מסקנה כפל של מספר חיובי בשלילי יתן תמיד מספר שלילי.

כפל מספר שלילי בשלילי

? מה יתקבל מכפל של שני מספרים שליליים:


$$(-3) * (-7)$$

בדוק במחשב! 

יתכן כי התוצאה קצת מפתיעה אותך - משני מספרים שליליים קיבלת מספר חיובי. אבל ראית כבר קודם כי פעולה רצופה של שני מינוסים שקולה ל-"+".

? מה יתקבל אם תפקוד על המחשב:

$$-8 * -4$$

בדוק במחשב! 

זכור! כפל של שני מספרים שליליים זה בזה נותן מספר חיובי.

הכנס סימנים מתאימים כדי לקבל ?

$$+1 = (+5) ___ (-4) ___ (-3) ___ (-6)$$

(אם אתה מתקשה - פנה לנספח ב' - סדר ביצוע פעולות חשבון.)

חילוק

הכללים החלים על חילוק הם כמו אלו של כפל:

• חילוק מספר חיובי בשלילי (או להיפך) נותן תמיד מספר שלילי.

• חילוק שני מספרים שליליים נותן תמיד מספר חיובי.

? מה התוצאה של התרגיל: $(-6) / (-2) * (-3)$

בדוק עצמך במחשב!

? מה התוצאה של התרגיל: $(-6) / (-2 * -3)$

בדוק עצמך במחשב!

חזקה

? מה יתקבל אם תפקוד:

$$(-3) \uparrow 2$$

וזה לא צריך להפתיע כי:

$$(-3) \uparrow 2 = (-3) * (-3)$$

וכפל של שני מספרים שליליים נותן מספר חיובי.

? ומה במקרה הבא: $(-3) \uparrow 3$

בדוק עצמך במחשב. 

לסכום שתול פעולות חשבון מתאימות כדי לקבל את התוצאות

הרשומות:

$$-3\emptyset = (-3) ___ (-8) ___ (-7) ___ (+15)$$

$$2\emptyset = (-2\emptyset) ___ (+2) ___ (+5) ___ (-4)$$

נספח ב

סדר ביצוע פעולות חשבון

אתה כבר יודע כי המחשב יודע לבצע 5 פעולות חשבון:

1. החיבור: "+"
2. החיסור: "-"
3. הכפל: "*"
4. החילוק: "/"
5. והחזקה: "↑"

? מה לדעתך ידפיס המחשב אם תפקוד עליו לבצע את התרגיל הבא:

PRINT 2+3*4

20 או 14?

בסה! 

בתגובה המחשב הדפיס את המספר ____.

? איזו פעולה חשבונית הוא ביצע קודם: $3*4$ או $2+3$?
ובכן, כדי להגיע לתוצאה 14 המחשב היה חייב לבצע תחילה את
פעולת הכפל: $3*4=12$ ואחר-כך את פעולת החיבור: $2+12=14$.
(אם הוא היה מבצע תחילה את פעולת החיבור היינו מקבלים את
התוצאה: 20.)

מסקנה! כאשר המחשב מבצע תרגיל חשבוני הכולל פעולות כפל
וחיבור הוא תמיד יבצע את פעולות הכפל תחילה ורק לאחר מכן
את פעולות החיבור. ואפילו פעולת הכפל באה לאחר פעולת
החיבור בתרגיל.

? איזה תוצאה תתקבל לאחר שהמחשב יבצע את התרגיל הבא:

$3+4*3+5*3$

בדוק עצמך במחשב! 

?

ומה לגבי פעולת החילוק?

בוא נבדוק את התרגיל הבא: $3+3/3$

האם יתקבל, לדעתך, 2 או 4?

בדוק זאת במחשב.

מסקנה כאשר המחשב מבצע תרגיל בו משולבות פעולות חילוק

וחיבור הוא תמיד יבצע את פעולת החילוק ואפילו היא באה

אחרי פעולת החיבור בתרגיל.

חקור את פעולת החיסור

?

מה קורה כאשר בתרגיל משולבות פעולות חיסור וכפל?

פקוד על המחשב לבצע תרגיל הכולל כפל וחיסור וקבע איזו

פעולה המחשב ביצע תחילה: כפל או חיסור.

מסקנה אם ביצעת את חקירותיך בדקדקנות בודאי הגעת למסקנה

כי בתרגיל בו משולבות פעולות כפל וחיסור המחשב תמיד יבצע

את פעולת הכפל ללא קשר למקומה בתרגיל. וכך גם לגבי

החילוק – הוא תמיד קודם לחיסור.

שילוב פעולות כפל וחילוק

לפי איזה סדר מבצע המחשב תרגיל בו משולבות פעולות כפל

וחילוק?

פקוד על המחשב לבצע את שני התרגילים הבאים:

$$15/5*6 = \underline{\hspace{2cm}}$$

$$5*6/15 = \underline{\hspace{2cm}}$$

איזו פעולה ביצע המחשב תחילה בתרגיל הראשון? ____ (כפל, חילוק)

איזו פעולה ביצע המחשב תחילה בתרגיל השני? ____ (כפל, חילוק)

מסקנה: כאשר פעולות כפל וחילוק משולבות באותו תרגיל המחשן מבצע אותם לפי סדר הופעתן בתרגיל.
בתרגיל הראשון הוא ביצע תחילה את פעולת החילוק ולכן התקבל התוצאה: 18.
בתרגיל השני הוא ביצע תחילה את פעולת הכפל ולכן התקבלה התוצאה: 2.

? ומה קורה כאשר פעולות חיבור וחסור משולבות באותו תרגיל? כמו בחילוק וכפל כך גם חיבור וחסור - המחשב מבצע אותם לנ סדר הופעתן בתרגיל.

תרגיל

שתול פעולות חשבון מתאימות בין ארבעת המספרים הבאים:

$$4 \quad _ \quad 5 \quad _ \quad 3 \quad _ \quad 7 = 12$$

כך שתתקבל התוצאה 12.

בדוק עצמך במחשב. 

וכך גם לגבי התרגיל הבא:

$$6 \quad _ \quad 3 \quad _ \quad 3 \quad _ \quad 3 = 12$$

בדוק עצמך במחשב. 

חקור את החזקה:

? איזו תוצאה תתקבל כתוצאה מביצוע התרגיל הבא:

$$2*3+2$$

18 או 36?

בדוק זאת במחשב! 

מסקנה: פעולת החזקה תמיד קודמת לפעולת הכפל.

האם אתה יכול לשער מה יהיה לגבי החילוק?

בדוק עצמך בעזרת המחשב. 

תרגיל: האם תוכל לשער מה תהיה תוצאת התרגיל הבא:

$$2+3*8/2\uparrow 2-8 = \underline{\quad? \quad}$$

בדוק עצמך במחשב!

נסכם: כאשר בתרגיל מסוים משולבות פעולות חזקה, כפל, חילוק, חיבור וחסור,

יבצע המחשב תחילה את פעולת ה_____.

לאחר מכן לפי סדר הופעתן את פעולות ה_____ וה_____.

ולבסוף לפי סדר הופעתן את פעולות ה_____ וה_____.

פעולות עם סוגריים

מה תהיה התוצאה של התרגיל הבא:

$$(2+3)*5$$

בדוק במחשב!

מסקנה כאשר מופיעים סוגריים בתרגיל, המחשב מבצע תחילה את

כל פעולות החשבון אשר בתוך הסוגריים!

האם תוכל לשער מה יקרה בתרגיל הבא: ?

$$((1+3)*2)\uparrow 2$$

במקרה זה המחשב ביצע תחילה את פעולת ה_____ לאחר מכן

את פעולת ה_____ ולבסוף את פעולת ה_____.

? "שתול" סוגריים במקומות המתאימים כדי לקבל את התוצאה:

$$2 + 9/3 * 4 \uparrow 2 = 196$$

$$3 + 1 \uparrow 2 * 5 = 80$$

$$4 + 16/4 - 2 * 4 = 12$$

תשובות

תשובה 1 מעמוד 7

יש להוסיף

5 GRAPHICS

תשובה 2 מעמוד 13

10 GRAPHICS : CLS : BORDER 0

20 PLOT X, 0

30 X=X+1 : GO TO 20

תשובה 3 מעמוד 14

הציר האופקי הראשי מורכב מ-320 פיקסלים: מ-0 עד 319.
לכן המחשב לא ביצע את הפקודה: 0, 320 PLOT.

תשובה 4 מעמוד 14

10 GRAPHICS : CLS : BORDER 0

20 FOR X=0 TO 319

30 PLOT X, 0

40 NEXT X

50 FOR Y=0 TO 199

60 PLOT 0, Y

70 NEXT Y

תשובה 5 מעמוד 15

10 GRAPHICS : CLS : BORDER 0

20 FOR X=0 TO 319

30 PLOT X, 0

40 PLOT 0, X*199/319

50 NEXT

```
10 GRAPHICS : CLS
20 FOR X=80 TO 180
30 PLOT X, 50
40 PLOT X, 150
50 PLOT 80, X-30
60 PLOT 180, X-30
70 NEXT X
```

תשובה 7 מעמוד 15

```
10 GRAPHICS : CLS : A=100 : B=200
20 FOR Y=20 TO 70
30 FOR X=A TO B
40 PLOT X, Y
50 NEXT X
60 A=A+1 : B=B-1
70 NEXT Y
```

שים לב: הקפד לרשום

רווח בין A לבין "TO"!

בהמשך תבין מדוע...

תשובה 8 מעמוד 16

```
10 GRAPHICS : CLS
20 PLOT INT (RND(1)*320),INT (RND(1)*200)
30 GO TO 20
```

תשובה 9 מעמוד 16

נראה, למשל, מה קורה למשתנה Y: הוא מתחיל מ-0 וגדל עד שהוא מגיע ל-199. כאן המחשב מקבל הוראה, בשורה 50 לתת ל-Y את הערך 199-. בהמשך הוא מבצע:

$$ABS(-199+1) = ABS(-198) = 198$$

ולכן הקו בשבר ומתחיל לרדת.

10 GRAPHICS : CLS

הגרלה [20 X1 = INT (RND(1)*305)
30 Y1 = INT (RND(1)*185)

ציור [40 FOR A=0 TO 15
50 PLOT X1+A, Y1:PLOT X1+A, Y1+15:PLOT
ריבוע X1, Y1+A:PLOT X1+15, Y1+A
קטן 60 NEXT A

65 GET A\$: IF A\$="" THEN 65

70 TEXT : PRINT"{CLR HOME}" : INPUT "X=";
X:INPUT "Y="; Y

80 GRAPHICS : PLOT X, Y

המתנה 90 GET A\$: IF A\$="" THEN 90

100 RUN

תשובה 11 מעמוד 20

1020 PRINT AT 0, 24 ; "C"

1030 PRINT AT 39, 24 ; "D"

תשובה 12 מעמוד 20

FOR X=ATO 10:PRINT X:NEXT
אתה כתבת:

AT

המחשב חשב שכאן כתוב AT

אם תכתוב רווח בין ה-A לבין ה-TO - הכל יבוא על מקומו בשלום...

תשובה 13 מעמוד 21

יש להוסיף או לשנות:

70 PRINT AT 0,0; "X=":INPUT X:PRINT AT 2,0; X

75 PRINT AT 0,2; "Y=":INPUT Y:PRINT AT 2,2; Y

כמו כן יש למחוק את 65.

תשובה 14 מעמוד 21

65 PRINT AT 0,20;"(";X1;"",Y1;")"

תשובה 15 מעמוד 22

```
10 GRAPHICS : CLS
20 FOR X=0 TO 319
30 PLOT X, 100
40 PLOT OVER 1; X, 100
50 NEXT X
```

תשובה 16 מעמוד 23

```
10 GRAPHICS : CLS
20 FOR X=5 TO 319
30 PLOT X, 100
40 PLOT OVER 1; X-5, 100
50 NEXT X
```

פיקסל מימין
מחיקה משמאל

תשובה 17 מעמוד 27

```
10 GRAPHICS : PAPER 4 : CLS
```

יש לשנות:

תשובה 18 מעמוד 28

```
10 GRAPHICS
20 FOR C=0 TO 15
30 BORDER C
40 PAPER 15-C : CLS
50 FOR T=1 TO 250 : NEXT
60 NEXT C
70 GO TO 20
```

תשובה 19 מעמוד 29

```
70 PLOT INK 0 ; 126, 94
```

תשובה 20 מעמוד 29

```
80 PLOT INK 0; 120, 88
90 PLOT INK 0; 127, 88
100 PLOT INK 0; 127, 95
```

תשובה 21 מעמוד 30

(7, 20)

המשבצת היא

תשובה 22 מעמוד 30

120 PLOT INK 7 ; PAPER 7; 100, 90

תשובה 23 מעמוד 31

150 PLOT INK 4; 108, 90

תשובה 24 מעמוד 31

כל הנקודות שבתוך המשבצות הצבעוניות מופיעות בצבע לבן.
זה הצבע שבו המחשב מדליק את הפיקסלים - אלא אם כן תפקוד עליו
צבע דיו אחר!

תשובה 25 מעמוד 31

70 PLOT PAPER K; INK K; X, Y : יש לשנות:

תשובה 26 מעמוד 31

10 GRAPHICS : CLS

20 FOR Y=0 TO 24

30 FOR X=0 TO 39

40 PRINT PAPER K; AT X, Y; "□"

50 K=K+1 : IF K=15 THEN K=0

60 NEXT X, Y

שים לב - אפשר גם כך ←

תשובה 27 מעמוד 33

10 GRAPHICS : CLS : X=160 : Y=100

30 PLOT X, Y

40 GET A\$

50 IF A\$ = "↗" THEN X=X+1

60 IF A\$ = "↖" THEN X=X-1

70 IF A\$ = "↑" THEN Y=Y+1

80 IF A\$ = "↓" THEN Y=Y-1

90 GO TO 30

תשובה 28 מעמוד 34

50 X=X-(A\$="→"¹) + (A\$="←"¹) : יש לשנות:

60 Y=Y-(A\$="↑"¹) + (A\$="↓"¹)

ולמחוק את שורות 70 ו-80.

תשובה 29 מעמוד 34

30 PLOT INVERSE K; X, Y : יש להוסיף או לשנות:

70 IF A\$ = "-" THEN K=1

80 IF A\$ = "+" THEN K=0

תשובה 30 מעמוד 36

20 DIM X(400), Y(400) : יש לשנות או להוסיף:

40 GET A\$: IF A\$ = "" THEN 40

45 X(N)=X : Y(N) = Y:N = N+1

תשובה 31 מעמוד 36

85 IF A\$ = "*" THEN 200 : יש להוסיף:

200 CLS

210 FOR A=20 TO 300 STEP 40

220 FOR B=20 TO 200 STEP 40

230 FOR I=0 TO N-1

240 PLOT X(I)-160+A, Y(I)-100+B

250 NEXT I, B, A

תשובה 32 מעמוד 37

87 IF A\$ = "P" THEN 300 : יש להוסיף:

300 CLS : FOR I=0 TO N-1

320 PLOT (X(I)-160)*2+160, (Y(I)-100)*2+100

330 NEXT I

תוספת זו לתוכנית גורמת, בלחיצה על מקש "P", להגדלת הציור

פי-12!

תשובה 33 מעמוד 37

יש לשנות: 30 PLOT X,Y : PLOT 320-X, Y :

PLOT 320-X, 200-Y : PLOT X,200-Y

תשובה 34 מעמוד 38

המחשב הטמין, בפיקסל אקראי על המסך, מוקש בלתי נראה. משימתך היא להתקרב אליו ככל האפשר.

בעזרת מקשי-המסמן אתה מתחיל לבוע מנקודת הראשית, בערך. ככל שתתקרב אל המוקש הבלתי-נראה, צבעי המשבצות ילכו ויתקרבו לצבע שחור והצליל ילך ויעלה.

כאשר אתה סבור שהתקרבת למוקש ככל שאתה יכול - לחץ על המקש "+". אז יראה לך המחשב את המוקש כפיקסל ואת מיקומך כנקודה עבה המורכבת מ-4 פיקסלים צמודים!

תשובה 35 מעמוד 40

יש להשלים: 30 DRAW 319-40, 199-50

כמובן שאפשר גם: 30 DRAW 279, 149

תשובה 36 מעמוד 40

יש להוסיף: 40 PLOT 40,50 : DRAW -40, 149

תשובה 37 מעמוד 40

יש להוסיף: 50 PLOT 40, 50 : DRAW -40, -50

60 PLOT 40, 50 : DRAW 319-40, -50

תשובה 38 מעמוד 41

10 GRAPHICS : CLS

20 X=160 : Y=100

30 PLOT X, Y : DRAW-X, -Y

40 PLOT X, Y : DRAW-X, 199-Y

50 PLOT X, Y : DRAW 319-X, 199-Y

60 PLOT X, Y : DRAW 319-X, -Y

70 GET A\$: X=X-5*(A\$="1→")+5*(A\$="1←")

80 Y=Y-5*(A\$="1↑")+5*(A\$="1↓")

90 GO TO 30

```

10 GRAPHICS : CLS
20 FOR X=0 TO 319 STEP 2
30 PLOT X, 0 : DRAW 0, 199
40 NEXT X

```

תשובה 40 מעמוד 41

```

50 FOR Y=0 TO 199 STEP 2
60 PLOT 0, Y : DRAW 319, 0
70 NEXT Y

```

יש להוסיף:

תשובה 41 מעמוד 42

```

15 OVER 1 : K=2
20 FOR X=0 TO 319 STEP K
50 FOR Y=0 TO 199 STEP K
80 K=K+1 : GO TO 20

```

יש להוסיף או לשנות:

תשובה 42 מעמוד 42

```

10 GRAPHICS : CLS
20 FOR X=0 TO 319 STEP 3
30 PLOT 160, 100 : DRAW X-160, 99
40 NEXT X

```

תשובה 43 מעמוד 42

```

50 FOR Y=199 TO 0 STEP -3
60 PLOT 160, 100 : DRAW 159, Y-100
70 NEXT Y
80 FOR X=319 TO 0 STEP -3
90 PLOT 160, 100 : DRAW X-160, -100
100 NEXT X
110 FOR Y=0 TO 199 STEP 3
120 PLOT 160, 100 : DRAW -160, Y-100
130 NEXT Y

```

יש להוסיף:

תשובה 44 מעמוד 43

```
10 GRAPHICS : CLS
20 DRAW 0,199 : DRAW 319,0 : DRAW 0, -199 :
   DRAW -319, 0
```

תשובה 45 מעמוד 43

```
10 GRAPHICS : CLS : K=2
15 FOR N=0 TO 100 STEP 8
17 PLOT N, N
20 DRAW PAPER K; 0, 199-2*N
30 DRAW PAPER K; 319-2*N, 0
40 DRAW PAPER K; 0, -199+2*N
50 DRAW PAPER K; -319+2*N, 0
60 K=K+1 : NEXT N
```

תשובה 46 מעמוד 43

```
10 GRAPHICS : CLS : OVER 1
15 FOR N=0 TO 70 STEP 3
20 PLOT 30+N, 170-N
30 DRAW 20+N, 0 : DRAW 0, -20-N : DRAW
   -20-N, 0 : DRAW 0, 20+N
40 NEXT N
```

תשובה 47 מעמוד 43

```
10 GRAPHICS : CLS
20 FOR X=50 TO 150
30 PLOT X, 40 : DRAW 0, 50
40 NEXT X
```

תשובה 48 מעמוד 43

```
35 PLOT X, 50 : DRAW OVER           יש להוסיף:
   -(X>60 AND X<140) ; 0, 30
```

```

10 GRAPHICS : CLS : OVER 1
20 A=150 : B=160 : Y=100
30 FOR N=1 TO 17
40 FOR X=A TO B !TO לבין A
50 PLOT X,Y : DRAW 0, B-A
60 NEXT X
70 A=A-5 : B=B+5 : Y=Y-5
80 NEXT N

```

תשובה 50 מעמוד 45

```

10 GRAPHICS : CLS
20 FOR N=1 TO 5
30 X=INT (RND(1)*39)
40 Y=INT (RND(1)*24)+1
50 PRINT PAPER 5; AT X,Y; N:PRINT PAPER 5;
   AT X,Y-1; "   "
100 NEXT N
120 PRINT AT 0,0; "A=" : INPUT A :
   PRINT AT 2,0; "     " : PRINT AT 2,0; A
125 PRINT AT 0,2; "B=" : INPUT B :
   PRINT AT 2,2; "     " : PRINT AT 2,2; B
130 DRAW OVER 1; A, B
140 GO TO 120

```

תשובה 51 מעמוד 45

```

10 GRAPHICS : CLS : OVER 1
20 X=INT (RND(1)*319)-159
30 Y=INT (RND(1)*199)-99
40 PLOT 160, 100 : DRAW X, Y
50 GO TO 20

```

תשובה 52 מעמוד 45

```
10 GRAPHICS : CLS : OVER 1 : A=160 : B=100
20 PLOT A, B
30 X=INT (RND(1)*100)-50 : Y=INT (RND(1)*100)
  -50
40 A=A+X : B=B+Y
50 IF A>319 OR A<0 OR B>199 OR B<0
  THEN A=A-X : B=B-Y : GO TO 30
60 DRAW X,Y : GO TO 30
```

תשובה 53 מעמוד 47

$3*3+4*4 = 25$

אורך היתר הוא 5 כי

$5*5 = 25$ וכידוע

תשובה 54 מעמוד 47

83.2586333

אורך היתר הוא

תשובה 55 מעמוד 47

PRINT SQR (30)

בתגובה ל

המחשב כותב 5.47722558 וזהו אורך הצלע A.

תשובה 56 מעמוד 48

```
10 GRAPHICS : CLS : OVER 1
20 FOR X=0 TO 199 STEP 5
30 PLOT X, 0 : DRAW -X, SQR (199*199-X*X)
40 NEXT X
```

תשובה 57 מעמוד 48

הרדיוס של מעגל זה הוא 50 ומרכזו נמצא ב-(150, 100).

תשובה 58 מעמוד 49

```
10 GRAPHICS : CLS
20 XC=160 : YC=100 : R=90 : PLOT XC, YC
30 INPUT A          -R עדי R-
40 PLOT XC+A, YC+SQR (R*R-A*A)
50 GO TO 30
```


תשובה 59 מעמוד 49

30 FOR A=-R TO R יש לשנות:

50 NEXT A

תשובה 60 מעמוד 49

45 PLOT XC+A, YC-SQR(R*R-A*A) : יש להוסיף:

תשובה 61 מעמוד 50

45 DRAW Ø,-2*SQR (R*R-A*A) :יש לשנות:

תשובה 62 מעמוד 50

25 R=20 : OVER 1 ישי להוסיף:

60 R=R+10 : GO TO 30

תשובה 63 מעמוד 50

40 PLOT XC+A, YC+2*SQR(R*R-A*A) : יש לשנות:

```
45 DRAW Ø, -4*SQR (R*R-A*A)
```

תשובה 64 מעמוד 50

40 PLOT XC+A,YC+SQR : יש לשבות או להוסיף:
(R*R-A*A)

```
45 PLOT XC+A, YC-SQR (R*R-A*A)
```

```
60 R=R+10 : XC=XC+10 : GO TO 30
```

תשובה 65 מעמוד 51

30 FOR A=-R TO 0 יש להוסיף או לשנות:

```
47 PLOT XC-A, YC+SQR (R*R-A*A)
```

48 PLOT XC-A, YC-SQR (R*R-A*A)

ולמחוק את 60.

תשובה 66 מעמוד 51

10 GRAPHICS : CLS : OVER 1

$$20 \quad XC=160 : YC=1000 : R=90$$

```
30 FOR A=-R TO R STEP 2
```

40 B=SQR (R*R-A*A)

```
50 PLOT XC, YC : DRAW -A,B : DRAW Ø,-2*B :
```

DRAW A, B

60 NEXT A

רשת המסך הגרפי

עמודות →

שורות ↓

0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20	21	22	23	24	25	26	27	28	29	30	31	32	33	34	35	36	37	38	39
0	8	16	24	32	40	48	56	64	72	80	88	96	104	112	120	128	136	144	152	160	168	176	184	192	200	208	216	224	232	240	248	256	264	272	280	288	296	304	312
7	15	23	31	39	47	55	63	71	79	87	95	103	111	119	127	135	143	151	159	167	175	183	191	199															

→ הקואורדינטות האופקיות של הפיקסלים

→ הקואורדינטות האנכיות של הפיקסלים

עד כה יצאו סדרות "מחשבת" עבור:

● מחשב COMMODORE 64

(3 החוברות הראשונות מתאימות גם ל־VIC-20)

● מחשבי SINCLAIR (SPECTRUM, ZX-81)

● מחשבי APPLE (II, II+, IIe, IIC)

● מחשבי ATARI (800XL, 600XL, 130XE)

● מחשב SPECTRAVIDEO

● מחשב AMSTRAD

● מחשב IBM-PC

● מחשב COMMODORE 128

● מחשב ATARI ST

יצא לאור "שעשועי מחשבת-1"

"שעשועי מחשבת-1" – מציינת את

תחילתה של סדרה חדשה של

"מחשבת" ומיועדת לכל חובבי

המחשבים הביתיים המחפשים אתגרים

תכנותיים.

מה תמצא בחוברת הראשונה של

"שעשועי מחשבת"?

הצענות מפורטות לפיתוח 5 משחקי

מחשב המתאימים מאד לתכנות בשפת

BASIC.

החוברת אינה כוללת אף תוכנית – את

זאת אנו משאירים לך לפתח.

בנוסף לכך תוכל להשתתף בהגרלה

נושאת פרסים של "מחשבת-BUG".

(פרסים תמצא בחוברת עצמה).



משהו על יחידה 4

ביחידת לימוד זאת הנקראת

"גרפיקה עדינה"

תלמד לפתח תוכניות בתחום הגרפיקה העדינה

כדי לעשות זאת עליך יהיה להשתמש

בקסטה של "מחשבת" – **HIGH RESOLUTION**,

שפותחה במיוחד עבור יחידת לימוד זו

ועבור היחידות שיבואו בעקבותיה.

נלמד אותך את כל הכלים התכנותיים

הדרושים לפיתוח התוכניות וכן נעשיר

את עולם המושגים המתמטי שלך על מנת

לפתח תוכניות מגוונות.

יחידת לימוד זאת, כמו הקודמות לה, בנויה כך,

שהיא תהווה עבורך אתגר מחשבתי ותכנותי

גם אם אתה יודע BASIC.

שים לב!

אתה חייב להיות מצויד בקסטה:

HIGH RESOLUTION

של "מחשבת" כדי ללמוד את

חוברת 4

מחשבת מ.ל. בע"מ – מערכות למידה